

INMC

International Nascom Microcomputer Club

BASIC programs BOOK I

CONTENTS =====

B1	HELLO	adapted by D. R. Hunt
B2	RUSSIAN ROULETTE	adapted by D. R. Hunt
B3	STARTREK	by R. Beal and others
B4	CUBIST ART	by D. R. Hunt
B5	3D NOUGHTS & CROSSES	adapted by R. Beal
B6	CALENDAR	by D. R. Hunt
B7	MAGIC LABYRINTH	adapted by H. Birkett
B8	ELIZA	adapted by D. R. Hunt
B9	CAMEL	adapted by D. R. Hunt
B10	COMRADE X	adapted by D. R. Hunt
Plus	SUB-ROUTINES	by Various
Appendix	Hints, reserved words etc.	

INMC
c/o Nascom Microcomputers Ltd.,
92, Broad Street,
Chesham, Bucks.

March 1980

Welcome to the first INMC book of BASIC programs. This book is the end result of the fun that members of the INMC have had with their Nascoms using the Nascom 8K BASIC on Nascom 1 or 2 computers over the last few months. Most of the programs included in this edition are games, but as exercises in learning how to use the machines they are invaluable. Don't just type the program in and make it work, but study the programs to understand HOW they work.

In many instances, the programs are based on ideas from books, magazines and other sources (where these are known, they are acknowledged), proving that the contributors are not particularly innovative. However, the programs have usually had to be extensively rewritten, either to make them compatible with the Nascom screen format, or to add features that were not included in the originals, or, in a couple of instances, because they didn't work anyway.

Typing in the programs
=====

All the programs have been 'LISTed' from programs running on a Nascom, and although not guaranteed bug free (or particularly tidy,) they all run. Specifically, the programs were run on a 32K Nascom 2, running at 4MHz with wait states, fitted with NAS-SYS 1, NAS-CPA 1.3 and Nascom ROM Basic V4.7. In view of these specifics, certain changes will have to be made to the programs when running them on Nascoms with different specifications. For instance, delay loops will have to be halved for use at 2MHz, USR functions will have to be changed (if not automatically sensed in the program) if NAS-BUG is in use, and various PEEK, DEEK, POKE and DOKE statements will be totally wrong if Nascom Tape Basic is in use. In general these programs may be run on different machines using different BASICS, although a certain amount of rewriting will be inevitable.

It is possible to make the programs shorter by not typing in the REM statements. However, we can not guarantee that jumps are not made to REM lines as these often precede a subroutine for instance. Therefore care should be taken when omitting lines. In one instance (Startrek) the program with its arrays is so long that any attempt to 'space out' the text in an attempt to make it more readable will result in an OM Error when used with a 16K Nascom.

The INMC would like to thank the contributors of the programs, all of which have been submitted without payment (save the ego booster of seeing their names in print). All profits from the sale of this book will be ploughed back into the INMC to improve the facilities the INMC is able to offer.

Published by the:
International Nascom Microcomputer Club, Chesham, Bucks.
Copyright (C) INMC 1980.

```
10 REM          ***** HELLO *****
20 REM
30 REM  Adapted for Nascom 1/2 fitted with
40 REM  NAS-SYS 1 or 'T' series monitors, by
50 REM  D. R. Hunt   June 1979 ...
60 REM  from the program by David Ahl, in
70 REM  '101 Basic Computer Games' published
80 REM  by Creative Computing.
90 REM **
100 REM ** Clear screen & put up title
110 GOSUB 1440
120 REM **
130 REM ** Start game
140 PRINT
150 PRINT"Hello there, I'm a Nascom Computer"
160 PRINT:N$=""
170 INPUT"What is your first name please ";N$
180 IF N$<>"" THEN 200
190 PRINT:PRINT"Come on, don't be shy.":GOTO 170
200 N1$="":N2$=""
210 FOR I=1 TO LEN(N$):A=ASC(MID$(N$,I,1))
220 IF I=1 OR I=1 THEN IF A>=96 THEN A=A-32
230 IF I=1 OR I=1 THEN 260
240 IF A<=57 THEN 260
250 IF A<=95 THEN A=A+32
260 N1$=N1$+CHR$(A)
270 I=0:IF A=32 THEN Y2$="":I=1:GOTO 310
280 Y2$=Y2$+CHR$(A)
290 IF A<=57 THEN I=1
300 IF Y2$="Ic" THEN I=1
310 NEXT:N$=N1$
320 PRINT:PRINT"Good day, are you enjoying ";
330 PRINT"yourself here":PRINT"today ";
340 PRINT N$;:B$=" ":INPUT" ";B$:PRINT
350 IF B$="YES" OR B$="yes" THEN 410
360 IF B$="NO" OR B$="no" THEN 430
370 PRINT"Sorry, I don't understand your ";
380 PRINT"answer of":PRINT"";B$;"". Please ";
390 PRINT"answer with a yes or no "
400 GOTO 340
410 PRINT"I'm glad to hear that ";N$;". "
420 GOTO 460
430 PRINT"I'm sorry to hear that ";N$;". "
440 PRINT"Maybe I can be of some help and ";
450 PRINT"cheer you up."
460 PRINT
470 PRINT"You know ";N$;" I can solve all "
480 PRINT"kinds of problems except those ";
490 PRINT"dealing with":PRINT"Greece. What ";
500 PRINT"kinds of problems do you have"
510 PRINT"(Answer sex, health, money, job) ";
520 C$=" "
530 INPUT C$:GOSUB 1440:PRINT:PRINT:PRINT
540 IF C$="SEX" OR C$="sex" THEN 830
550 IF C$="HEALTH" OR C$="health" THEN 760
560 IF C$="MONEY" OR C$="money" THEN 700
```

```
570 IF C$="JOB" OR C$="job" THEN 610
580 PRINT"I'm sorry ";N$;" but your answer of"
590 PRINT"";C$;" is all Greek to me."
600 GOTO1040
610 PRINT"I can sympathize with you ";N$;". "
620 PRINT"I have to work nine hours a day ";
630 PRINT"six days a week":PRINT"and with no ";
640 PRINT"pay .... and some of my bosses "
650 PRINT"really beat my keyboard like it was";
660 PRINT" my fault.":PRINT"My advice to you ";
670 PRINT N$;" is to open a":PRINT"retail ";
680 PRINT"computer store, it's great fun."
690 GOTO1040
700 PRINT"Sorry ";N$;", I'm broke too."
710 PRINT"Why don't you sell encyclopedias ";
720 PRINT"or marry":PRINT"someone rich, or ";
730 PRINT"stop eating, or something."
740 PRINT"That way you'll need less money."
750 GOTO1040
760 PRINT"My advice to you ";N$;" is to"
770 PRINT" 1 Take an aspirin."
780 PRINT" 2 Drink plenty of liquids (";
790 PRINT"orange juice,"
800 PRINT"      not beer)."
810 PRINT" 3 Go to bed (alone)."
820 GOTO1040
830 PRINT"Is your problem ";
840 PRINT"too much ?":D$=" "
850 INPUT "Or too little";D$
860 IF D$="TOO MUCH" OR D$="too much" THEN 920
870 IF D$="TOO LITTLE" THEN 970
880 IF D$="too little" THEN 970
890 PRINT"Come on don't let such a little ";
900 PRINT"thing embarrass":PRINT"you ";N$;
910 PRINT" which is it, ";:GOTO 840
920 PRINT"You call that a problem, I should ";
930 PRINT"have such":PRINT"problems. If it ";
940 PRINT"worries you that much "
950 PRINT N$;", try a cold shower."
960 GOTO1040
970 PRINT"Then why are you here ";N$;","
980 PRINT"you should be in Paris, or ";
990 PRINT"Hamburg, or Tokyo,"
1000 PRINT"or some place where there's some ";
1010 PRINT"real action."
1020 PRINT"Don't spend all day hanging around ";
1030 PRINT"here. Life's":PRINT"too short."
1040 PRINT
1050 PRINT"Do you want any more of your ";
1060 PRINT"problems solved":E$=" "
1070 PRINT N$;" ";:INPUT E$:PRINT
1080 IF E$="YES" OR E$="yes" THEN 1120
1090 IF E$="NO" OR E$="no" THEN 1140
1100 PRINT"Just a simple yes or no ";N$
1110 GOTO1050
1120 PRINT"What kind (sex, money, health, ";
```

```
1130 PRINT"job)";:GOTO 520
1140 GOSUB 1440:PRINT:PRINT:PRINT
1150 PRINT"That will be #5.00 for the advice ";
1160 PRINT N$"."
1170 PRINT"Please leave the money on the ";
1180 PRINT"keyboard."
1190 FOR I=1 TO 2000:NEXT
1200 PRINT:PRINT:PRINT:G$=" "
1210 INPUT"Did you leave the money ";G$
1220 IF G$="YES" OR G$="yes" THEN 1270
1230 IF G$="NO" OR G$="no" THEN 1340
1240 PRINT"Your answer of '";G$;"' confuses me."
1250 PRINT N$;" please respond with a yes or ";
1260 PRINT"no.":GOTO 1210
1270 PRINT"Hey, ";N$;" ??? You left no money ";
1280 PRINT"at all !"
1290 PRINT"You're cheating me out of my hard-";
1300 PRINT"living."
1310 PRINT"RIP OFF, ";N$;" you're a *****!!"
1320 PRINT"Patients like you I can get any day."
1330 GOTO 1420
1340 PRINT"Thats honest ";N$;" , but how do you"
1350 PRINT"expect a hard-working computer to ";
1360 PRINT"go on with"
1370 PRINT"its psychology studies ";
1380 PRINT"when the customers don't"
1390 PRINT"pay their bills."
1400 PRINT"Still, nice meeting you ";N$;". "
1410 PRINT"Have a nice day."
1420 PRINT"Now let me talk to someone else."
1430 FOR I=1 TO 10000:NEXT:GOTO 110
1440 CLS:T$="HELLO":FOR T=1 TO LEN(T$)
1450 POKE 3037+T,ASC(MID$(T$,T,1)):NEXT
1460 RETURN
```

Ok

Don't Delete When You Can Insert.

=====

Did you know that if you have missed a word (or several) out of a BASIC line it isn't necessary to type out the line again (provided you have a Nas-Sys, and you're mad if you haven't). Just move the cursor to the point in the line where you want to make the insert, and open the line up by using the "Shift" key in conjunction with the "Cursor Right" key

Similarly to delete the central portion of a line, use the "Shift" key with the "Cursor Left" key. It is in the manual, but we still get queries about it. Why not read the manuals again, you'll be amazed how much you've missed on the first 15 readings.

```
10 REM      ***** RUSSIAN ROULETTE *****
20 REM
30 REM  Adapted for Nascom 1/2 fitted with
40 REM  NAS-SYS 1 or 'T' series monitors.
50 REM  by D. R. Hunt.   June 1979
60 REM  from the program by Tom Adamtex, in
70 REM  '101 Basic Computer Games' published
80 REM  Creative Computing.
90 REM
100 REM ** Set up machine code input
110 DATA 31711,1080,-53,536,-20665,3370,-5664,0
120 DATA 27085,14336,-13564,6399,18178,10927,-8179,233
130 IF PEEK(1)<>0 THEN RESTORE 120
140 DOKE 4100,3328:FOR A1=3328 TO 3342 STEP 2
150 READ A2:DOKE A1,A2:NEXT
160 REM **
170 REM ** Start game
180 CLS
190 SCREEN 1,7:N$="":N1$="":PRINT"W";
200 INPUT"hat is your name ";N$
210 IF N$=<> "" THEN 230
220 SCREEN 1,7:PRINT"Please .. w";:GOTO 200
230 FOR I=1 TO LEN(N$):A=ASC(MID$(N$,I,1))
240 IF I=1 OR I=1 THEN IF A>=96 THEN A=A-32
250 IF I=1 OR I=1 THEN 280
260 IF A<=57 THEN 280
270 IF A<=95 THEN A=A+32
280 N1$=N1$+CHR$(A)
290 I=0:IF A=32 THEN N2$="":I=1:GOTO 310
300 N2$=N2$+CHR$(A)
310 IF A<=57 THEN I=1
320 IF N2$="Mc" THEN I=1
330 NEXT:N$=N1$
340 GOSUB 950
350 PRINT:PRINT:PRINT
360 PRINT "Ok ";N$;" you'll find this easy."
370 PRINT"You don't need instructions for thi";
380 PRINT"s. Here is a":PRINT"revolver it has";
390 PRINT" six chambers, one is loaded."
400 PRINT"You only have to survive 10 tries."
410 PRINT:PRINT"Type 'Y' for a spin."
420 PRINT"Type 'N' to give up.":N=0:J=0
430 PRINT"There is no need to 'hit' newline.
440 PRINT:PRINT"Your turn. Go ";
450 I=USR(0): IF I<0 THEN 450
460 PRINT CHR$(I): IF I=78 THEN 820
470 IF I=89 THEN 510
480 PRINT:PRINT"Didn't get that right did ";
490 PRINT"you ";N$;". "
500 PRINT"Nervous ? Try again !":GOTO 440
510 N=N+1: IF RND(1)>.833333 THEN 690
520 PRINT"      - CLICK -"
530 IF N=3 THEN 580
540 IF N=5 THEN 610
550 IF N=7 THEN 630
560 IF N=9 THEN 650
```

```

570 GOTO 660
580 PRINT"There, ";N$;" its not painful, ";
590 PRINT"is it."
600 GOTO 660
610 PRINT"Bet you're starting to sweat a bit."
620 GOTO 660
630 PRINT"Only a couple more to go now."
640 GOTO 660
650 PRINT"Last one now, can't lose really."
660 IF J=0 THEN 830
670 IF N>9 THEN 760
680 J=0:GOTO 440
690 PRINT:PRINT"          BANG !!!"
700 PRINT:PRINT"Sorry ";N$;" you're dead."
710 PRINT"Please lie on the floor and we ";
720 PRINT"will inform your":PRINT"next of kin."
730 FOR I=1 TO 10000:NEXT:REM 5000 for 2MHz
740 CLS:SCREEN 1,6
750 PRINT"Next victim please.":GOTO 190
760 PRINT:PRINT"Congratulations, well done ";
770 PRINT N$;". "
780 PRINT"You win. It wasn't difficult ";
790 PRINT"was it !!!"
800 PRINT"Now let someone else blow his ";
810 PRINT"brains out.":GOTO 730
820 PRINT:PRINT"CHICKEN !!!!": GOTO 800
830 FOR I=1 TO 500:NEXT
840 PRINT:PRINT"My turn.":N=N+1
850 IF RND(1)>.833333 THEN 870
860 PRINT"          - CLICK -":GOTO 670
870 PRINT:PRINT"          BANG !!!"
880 PRINT:PRINT"Oh dear, I've shot myself."
890 FOR I=1 TO 2000:NEXT:CLS:REM 1000 for 2MHz
900 SCREEN 16,7:PRINT"AGHHHHHHHHHHHHH!!!"
910 FOR I=1 TO 400:NEXT:CLS
920 FOR I=1 TO 20000:NEXT:REM 10000 for 2MHz
930 SCREEN 14,7:PRINT"Bet that fooled you."
940 FOR I=1 TO 2000:NEXT:CLS:GOTO 740
950 CLS:T$="RUSSIAN ROULETTE"
960 FOR T=1 TO LEN(T$)
970 POKE 3032+T,ASC(MID$(T$,T,1)):NEXT:RETURN
Ok

```

Shift That Cursor.

===== :=====

Did you know that all the screen handling capabilities of Nas-Sys can be used under the control of your BASIC program. Just look up the hex code of the function in the Nas-Sys manual (on the page entitled "Screen Editing"), convert it to decimal, and use it as the operand of a CHR\$ function.

For example, the code for Clear Screen is 0CH (12 decimal), so ?CHR\$(12) will have the same effect as CLS. All of the other codes can be used to useful effect.

```
10 REM          ***** STARTREK *****
20 REM
30 REM  Written for 16 K Nascom 1/2 fitted with
40 REM  NAS-SYS 1 or 'T' series monitors, by
50 REM  Richard Beal      June 1979.
60 REM  Minor alterations by D. R. Hunt.
70 REM
80 REM ** Set up machine code input
90 DATA25055,1080,-53,536,-20665,3370,-5664,0
100 DATA27085,14336,-13564,6339,18178,10927,-8179,233
110 IFPEEK(1)<>0THENRESTORE100
120 DOKE4100,3328:FORI=3328TO3342STEP2
130 READI1:DOKEI,11:NEXT
140 REM
150 REM ** Initialise
160 CLEAR250:CLS
170 DIMA(7,7),S$(7,7),D(6),KP(62),KX(62),KY(62)
180 DIMG(7,7),H(7,7)
190 PI=3.141593:M9=-9999:GOSUB5710
200 DEFFNR(X)=INT(X*100+.5)/100
210 S9=0:GOSUB4060:GOSUB5110:GOTO250
220 REM
230 REM ** Restart here.
240 S9=0:GOSUB4060
250 Q$="":INPUT"Do you want instructions ";Q$
260 IFQ$<>"YES"ANDQ$<>"NO"GOTO4120
270 IFQ$="YES"THENGOSUB4170
280 DF=2:Q$=""
290 INPUT"Do you want a difficult game ";Q$
300 IFQ$<>"YES"ANDQ$<>"NO"GOTO4140
310 IFQ$="YES"THENDF=1
320 Q5=-1:INPUT"Klingon mobility (0-100%) ";Q5
330 Q5=Q5/100:IFQ5>=0ANDQ5<=1GOTO350
340 PRINT"Please concentrate!":GOTO320
350 GOSUB4060:PRINT"Entering galaxy";
360 PRINTINT(RND(1)*1E5)
370 D=60+INT(RND(1)*10):D9=D
380 B=DF*2+INT(RND(1)*3):B9=B
390 K=(3-DF)*14+INT(RND(1)*10):K9=K
400 C=85+DF*20+INT(RND(1)*15):C9=C
410 FORI=0TO7:FORJ=0TO7:H(I,J)=0
420 A(I,J)=INT(RND(1)*4*(3-DF)):NEXT:NEXT
430 FORI=1TOB:GOSUB1200
440 A(X1,Y1)=A(X1,Y1)+10:NEXT
450 FORI=1TOK:GOSUB1200
460 A(X1,Y1)=A(X1,Y1)+100:NEXT
470 S9=3000+INT(RND(1)*1000)
480 F8=0:F7=0:F=15:GOSUB1150
490 IFF<KTHENF=K+INT(RND(1)*5)
500 PRINT"Stardate"S9
510 PRINT"Captain "N$":
520 PRINT"Your mission is to destroy"K;
530 PRINT"Klingons":PRINT" in"D"stardates."
540 PRINT"You have a crew of"C
550 PRINT"There are"B"starbases in this ";
560 PRINT"galaxy."
```



```
570 PRINT"You have"F"photon torpedoes."
580 REM
590 REM ** Random initial position.
600 GOSUB1200:U=X1:V=Y1
610 GOSUB1200:X=X1:Y=Y1
620 REM
630 REM ** Enter quadrant.
640 FORI=0TO7:FORJ=0TO7:S$(I,J)=X$
650 NEXT:NEXT
660 S$(X,Y)=E$:O=0
670 GOSUB4360
680 IFM1=0GOTO710
690 I$=K$:FORI=0TOM1-1:GOSUB1230:KP(I)=300
700 KX(I)=X1:KY(I)=Y1:NEXT
710 KP(M1)=M3
720 IFM2>0THENI$=B$:FORI=1TOM2:GOSUB1230:NEXT
730 IFM3>0THENI$=S$:FORI=1TOM3:GOSUB1230:NEXT
740 GOSUB1040
750 REM
760 REM ** Make moves, etc.
770 GOSUB5400:GOSUB2100
780 REM
790 REM ** Check for end.
800 IFD=0GOTO4750
810 IFE<=0GOTO4780
820 IFE>10000THENE=6000
830 IFC=0GOTO4810
840 REM
850 REM ** Get command.
860 Q$="":PRINT"Captain ";N$;"! Command ? ";
870 I7=4:GOTO5300
880 Q$=Q1$
890 IFQ$="H"ORQ$="HELP"GOTO4920
900 IFQ$="R"GOTO2660
910 IFQ$="S"GOTO1600
920 IFQ$="W"GOTO2930
930 IFQ$="L"GOTO1460
940 IFQ$="P"GOTO1710
950 IFQ$="T"GOTO3490
960 IFQ$="G"GOTO1320
970 IFQ$="D"GOTO2850
980 IFQ$="A"GOTO4830
990 IFQ$=""GOTO860
1000 PRINT"Captain, please give a valid order!"
1010 D1=.05:GOSUB5040:GOTO800
1020 REM
1030 REM ** Check if docked.
1040 FORI=X+(X>0)TOX-(X<7)
1050 FORJ=Y+(Y>0)TOY-(Y<7)
1060 IFS$(I,J)=B$GOTO1080
1070 NEXT:NEXT:O=0:RETURN
1080 IFO=1THENRETURN
1090 GOSUB1150
1100 PRINT"Sulu: 'Captain, we are docked at ";
1110 PRINT"the starbase.'"
1120 O=1:RETURN
```

```
1130 REM
1140 REM ** Restock.
1150 IFE>100000RE<6000THENE=6000
1160 IFF<15THENF=15
1170 FORI=0TO6:D(I)=0:NEXT:RETURN
1180 REM
1190 REM ** Random coordinate.
1200 X1=INT(RND(1)*8):Y1=INT(RND(1)*8):RETURN
1210 REM
1220 REM ** Put item into sector
1230 X1=INT(RND(1)*8):Y1=INT(RND(1)*8)
1240 IFS$(X1,Y1)<>X$GOTO1230
1250 S$(X1,Y1)=I$:RETURN
1260 REM
1270 REM ** Report position.
1280 PRINT"Enterprise is at Q"U;V" S"X;Y
1290 RETURN
1300 REM
1310 REM ** Galactic record.
1320 GOSUB4060:PRINT"Galactic Record:";
1330 PRINT" (Klingons, Starbases, Stars)"
1340 GOSUB1280:J5=1:GOSUB2530
1350 IFI5>0GOTO1010
1360 PRINT"      0      1      2      3      4";
1370 PRINT"      5      6      7":PRINT
1380 FORI=0TO7:PRINTI": ";
1390 FORJ=0TO7
1400 IFH(I,J)=0THENPRINTG$;:GOTO1420
1410 PRINTRIGHT$(" "+STR$(G(I,J)),5);
1420 NEXT:PRINT:NEXT
1430 D1=.06:GOSUB5040:GOTO800
1440 REM
1450 REM ** Long range scan.
1460 GOSUB4060:PRINT"Long Range Scan:";
1470 PRINT" (Klingons, Starbases, Stars)"
1480 GOSUB1280:J5=2:GOSUB2530
1490 IFI5>0GOTO1010
1500 PRINT"      "V-1"      "V"      "V+1:PRINT
1510 FORI=U-1TOU+1:PRINTI": ";
1520 FORJ=V-1TOV+1
1530 IFI<0ORJ>7ORJ<0ORJ>7THENPRINTL$;:GOTO1560
1540 PRINTRIGHT$(" "+STR$(A(I,J)),6);
1550 H(I,J)=1:G(I,J)=A(I,J)
1560 NEXT:PRINT:NEXT:PRINT
1570 D1=.1:GOSUB5040:E=E-20:GOTO800
1580 REM
1590 REM ** Short range scan.
1600 GOSUB4060:PRINT"Short Range Scan:"
1610 GOSUB1280:J5=0:GOSUB2530
1620 IFI5>0GOTO1010
1630 H(U,V)=1:G(U,V)=A(U,V)
1640 PRINT"      0 1 2 3 4 5 6 7"
1650 FORI=0TO7:PRINT"      "I;
1660 FORJ=0TO7:PRINT" "S$(I,J);:NEXT:PRINT
1670 NEXT
1680 D1=.1:GOSUB5040:E=E-10:GOTO800
```

```
1690 REM
1700 REM ** Phasor.
1710 J5=3:GOSUB2530:IF I5>0GOTO1010
1720 A=0
1730 PRINT"Phasor energised. How much energy ? ";
1740 I7=2:GOTO5300
1750 A=VAL(Q1$)
1760 A=INT(A):IFA<0GOTO1000
1770 IFA=0GOTO1010
1780 IFA<=EGOTO1810
1790 PRINT"Spock: 'We only have"E"units.'"
1800 GOTO1010
1810 E=E-A:IFM1>0GOTO1850
1820 PRINT"Phasor fired, but no Klingons!"
1830 IFK=0GOTO4420
1840 D1=.5:GOSUB5040:GOTO770
1850 A=INT(A/M1)
1860 FORM=0T062:IFKP(M)=M9GOTO1830
1870 IFKP(M)=0GOTO1900
1880 GOSUB1930:PRINTS"units hit. ";
1890 GOSUB2000
1900 NEXT:GOTO1830
1910 REM
1920 REM ** Calculate energy loss.
1930 IFA<=1000GOTO1960
1940 PRINT"Dilithium crystals overloaded."
1950 J5=3:D(J5)=D(J5)+1:GOSUB2530:A=1000
1960 I=KX(M)-X:J=KY(M)-Y
1970 S=INT(A*30/(30+I*I+J*J)+1):RETURN
1980 REM
1990 REM ** Damage to Klingon.
2000 PRINT"Klingon "CHR$(65+M)" at S";
2010 PRINTKX(M);KY(M);
2020 KP(M)=KP(M)-S
2030 IFKP(M)>0THENPRINT"damaged.":RETURN
2040 KP(M)=0:K=K-1:M1=M1-1
2050 A(U,V)=A(U,V)-100:G(U,V)=A(U,V)
2060 S$(KX(M),KY(M))=X$
2070 PRINT"destroyed.":RETURN
2080 REM
2090 REM ** Klingon attack.
2100 IFM1=0THENRETURN
2110 PRINT"KLINGON ATTACK!!!"
2120 IFO=0GOTO2280
2130 PRINT"Starbase shield. ";
2140 PRINT"Enterprise, but 't's under "
2150 PRINT"attack too."
2160 IFRND(1)>.15*SQR(M1)THENRETURN
2170 PRINT" *** BOOM ***"
2180 PRINT"Starbase destroyed!!!!"
2190 B=B-1:M2=M2-1
2200 A(U,V)=A(U,V)-10:G(U,V)=A(U,V)
2210 FORI=X+(X>0)TOX-(X<7)
2220 FORJ=Y+(Y>0)TOY-(Y<7)
2230 IFS$(I,J)=B$THENS$(I,J)=X$:GOTO2250
2240 NEXT:NEXT
```

```
2250 O=0:T=200:GOSUB2430:GOSUB1040:RETURN
2260 REM
2270 REM ** Damage by Klingon.
2280 T=0:FORM=0T062:IFKP(M)=N9GOTO2350
2290 IFKP(M)=0GOTO2340
2300 A=(KP(M)+KP(M)*RND(1))/2
2310 GOSUB1930:T=T+S
2320 PRINTS"units hit from Klingon "CHR$(65+I);
2330 PRINT" at S"KX(I);KY(I)
2340 NEXT
2350 E=E-T:IFE>0GOTO2380
2360 PRINT" *** BANG ***"
2370 FORI=1TO500:NEXT:RETURN
2380 PRINTE"energy units left."
2390 IF(E*RND(1)/6)<TGOTO2430
2400 PRINT"No damage.":RETURN
2410 REM
2420 REM ** Decide on damage.
2430 IFD(6)>0GOTO2470
2440 D(6)=INT(RND(1)*T/75):J5=6:GOSUB2530
2450 IFI5>0THEN2470
2460 PRINT"Shields prevented damage.":RETURN
2470 J5=INT(RND(1)*6)
2480 D(J5)=D(J5)+INT(RND(1)*T/100)+1
2490 I6=INT(RND(1)*T/50):IFI6=0GOTO2530
2500 C=C-I6:IFC<0THENI6=C:C=0
2510 PRINT"McCoy: 'Sickbay here, we ";
2520 PRINT"had" I6 "casualties.'"
2530 I5=D(J5):IFI5=0THENRETURN
2540 IFJ5=0THENPRINT"Short range sensor";
2550 IFJ5=1THENPRINT"Galactic record display";
2560 IFJ5=2THENPRINT"Long range sensor";
2570 IFJ5=3THENPRINT"Phasor";
2580 IFJ5=4THENPRINT"Warp engines";
2590 IFJ5=5THENPRINT"Photon torpedo tubes";
2600 IFJ5=6THENPRINT"Shields";
2610 PRINT" damaged.":PRINTI5;"stardates ";
2620 PRINT"estimated for repair."
2630 RETURN
2640 REM
2650 REM ** Status report.
2660 GOSUB4060:PRINT"Status Report:"
2670 PRINT"Stardate"FNR(S9+D9-D)
2680 PRINT"Captain "N$" in command."
2690 PRINT FNR(D9-D)"stardates";
2700 PRINT" used and"D"stardates left."
2710 GOSUB1280
2720 IFO=1THENPRINT"Docked at starbase."
2730 IFM1>0THENPRINT"** Condition RED ** ";
2740 IFM1>0THENPRINTM1"Klingons in quadrant."
2750 IFM1=0THENPRINT"Condition Green."
2760 IFE<1000THENPRINT"*** Energy low ***"
2770 PRINTE"energy units."
2780 PRINTC"crew."
2790 PRINTB"starbases."
2800 PRINTF"photon torpedoes."
```

```
2810 PRINTK"Klingons left to destroy."
2820 D1=.01:GOSUB5040:GOTO800
2830 REM
2840 REM ** Damage report.
2850 GOSUB4060:PRINT"Damage Report:"
2860 A=0:FORJ5=0TO6
2870 GOSUB2530:IFI5>0THENA=1
2880 NEXT:IFA>0GOTO2820
2890 PRINT:PRINT"All systems functioning."
2900 PRINT:GOTO2820
2910 REM
2920 REM ** Warp engines.
2930 J5=4:GOSUB2530
2940 IFI5>0THENPRINT"Impulse engines only."
2950 W=0:PRINT"Warp factor (in sectors) ? ";
2960 I7=1:GOTO5300
2970 W=VAL(Q1$)
2980 IFW<0GOTO1000
2990 IFW<1GOTO1010
3000 IFW<=3OR I5=0GOTO3030
3010 PRINT"Chekov: 'We can try 3 at the most.'"
3020 GOTO2950
3030 IFW<91GOTO3060
3040 PRINT"Spock: 'Are you sure, Captain?'"
3050 GOTO2950
3060 IFE>=INT(W*20)GOTO3110
3070 PRINT"Scotty: 'Insufficient energy, ";
3080 PRINT"Sir.'"
3090 PRINT"Maximum warp factor is"INT(E/20)
3100 GOTO2950
3110 GOSUB3920:IFR=0GOTO1010
3120 D1=1:GOSUB5040:E=E-INT(W*20)
3130 S$(X,Y)=X$:P=X:G=Y:I=X:J=Y:Z=0
3140 IFW<(R/2)GOTO3270
3150 W=W-R:O=0
3160 P=P+S:G=G+T:I=INT(P+.5):J=INT(G+.5)
3170 IFI<0THENZ=1:U=U-1:P=P+8:I=7
3180 IFI>7THENZ=1:U=U+1:P=P-8:I=0
3190 IFJ<0THENZ=1:V=V-1:G=G+8:J=7
3200 IFJ>7THENZ=1:V=V+1:G=G-8:J=0
3210 IFZ<>0GOTO3140
3220 IFS$(I,J)=X$THENX=I:Y=J:GOTO3140
3230 PRINT"*** EMERGENCY STOP ***"
3240 T=500:GOSUB2430
3250 S$(X,Y)=E$
3260 GOSUB1280:GOSUB1040:GOTO770
3270 IFZ=0THENX=I:Y=J:GOTO3250
3280 IFRND(1)>.1GOTO3310
3290 PRINT"*** SPACE STORM ***"
3300 T=300:GOSUB2430
3310 IFRND(1)>.1GOTO3370
3320 PRINT"You have plunged into a rotating ";
3330 PRINT"Black Hole!":PRINT"Energy gain of";
3340 A=INT(RND(1)*3000):E=E+A:PRINT A"units."
3350 PRINT"Ship hurtles through space warp!"
3360 PRINT"Position unknown!":GOTO600
```

```
3370 IFU<00RU>70RV<00RV>7GOTO3390
3380 X=I:Y=J:GOSUB1280:GOTO640
3390 PRINT"You have attempted to leave the ";
3400 PRINT"galaxy."
3410 PRINT"Curved space prevented this."
3420 IFU<0THENU=0
3430 IFU>7THENU=7
3440 IFV<0THENV=0
3450 IFV>7THENV=7
3460 GOTO3380
3470 REM
3480 REM ** Photon torpedo.
3490 J5=5:GOSUB2530:IFI5>0GOTO1010
3500 IFF>0GOTO3530
3510 PRINT"Chekov: 'No torpedoes left!'"
3520 GOTO1010
3530 PRINT"Photon torpedo loaded. ";
3540 GOSUB3920:IFR=0GOTO1010
3550 F=F-1:F8=F8+1:F7=F7+1
3560 PRINT"Tracking...";
3570 P=X:G=Y
3580 P=P+S:G=G+T:I=INT(P+.5):J=INT(G+.5)
3590 IFI<00RI>70RJ<00RJ>7GOTO3670
3600 PRINT"  "MID$(STR$(I),2,1)"-";
3610 PRINT MID$(STR$(J),2,1);
3620 IFS$(I,J)=X$GOTO3580
3630 PRINT
3640 IFS$(I,J)=K$GOTO3700
3650 IFS$(I,J)=B$GOTO3750
3660 IFS$(I,J)=S$GOTO3800
3670 PRINT:PRINT"It missed."
3680 IFK=0GOTO4420
3690 D1=.5:GOSUB5040:E=E-20:GOSUB1040:GOTO770
3700 F7=F7-1:S=INT(250+RND(1)*200)
3710 FORM=0T062:IFKP(M)=0GOTO3740
3720 IFKX(M)<>IORKY(M)<>JGOTO3740
3730 GOSUB2000:GOTO3680
3740 NEXT
3750 B=B-1:M2=M2-1
3760 A(U,V)=A(U,V)-10:G(U,V)=A(U,V)
3770 S$(I,J)=X$
3780 PRINT"Starbase destroyed."
3790 PRINT"Spock: 'Fascinating!':GOTO3680
3800 PRINT"It hit a star."
3810 IFRND(1)>.3GOTO3840
3820 PRINT"Photon torpedo absorbed by star."
3830 GOTO3580
3840 M3=M3-1:A(U,V)=A(U,V)-1:G(U,V)=A(U,V)
3850 S$(I,J)=X$:PRINT"Star destroyed."
3860 IFRND(1)>.5GOTO3680
3870 PRINT"It went nova!"
3880 PRINT"*** RADIATION ALARM ***"
3890 T=300:GOSUB2430:GOTO3680
3900 REM
3910 REM ** Get course.
3920 I=0:PRINT"Course (1-360) ? ";
```

```
3930 I7=3:GOTO5300
3940 I=VAL(Q1$)
3950 IF I>0AND I<=360GOTO3970
3960 R=0:RETURN
3970 S=INT((I+45)/90):I=TAN((I-S*90)*PI/180)
3980 IFS=4THENS=0
3990 R=SQR(1+I*I)
4000 IFS=0THENS=-1:T=I:RETURN
4010 IFS=1THENS=I:T=1:RETURN
4020 IFS=2THENS=1:T=-1:RETURN
4030 IFS=3THENS=-1:T=-1:RETURN
4040 REM
4050 REM ** Initial messages.
4060 CLS:H$=" *** STAR TREK ***"
4070 IFS9=0GOTO4090
4080 H$=H$+" Stardate"+STR$(FNR(S9+D9-D))
4090 FOR I=1TO LEN(H$)
4100 POKE I+3017,ASC(MID$(H$,I,1))
4110 NEXT:RETURN
4120 PRINT"Don't be silly, enter YES or NO."
4130 PRINT"I repeat. ";GOTO250
4140 PRINT"If you can't get it ";
4150 PRINT"together now, what will you ";
4160 PRINT"be like later? Try again!!":GOTO280
4170 GOSUB4060:PRINT"The Klingon Empire has ";
4180 PRINT"invaded the Federation."
4190 PRINT"You are Captain of the USS ";
4200 PRINT"Enterprise and your ";
4210 PRINT"mission is to seek out and ";
4220 PRINT"destroy the enemy."
4230 PRINT"The galaxy is divided into 64";
4240 PRINT" Quadrants (8*8),"
4250 PRINT"each of which contains 64 Sectors."
4260 PRINT"When asked for a command, you must";
4270 PRINT" enter the":PRINT"command letter."
4280 PRINT"For a list of commands, enter H ";
4290 PRINT"or HELP."
4300 PRINT"Angles:      360"
4310 PRINT"              270  +  90"
4320 PRINT"              180"
4330 PRINT:RETURN
4340 REM
4350 REM ** Find Klingons, bases, stars
4360 M1=INT(A(U,V)/100)
4370 M2=INT((A(U,V)-M1*100)/10)
4380 M3=A(U,V)-M1*100-M2*10
4390 RETURN
4400 REM
4410 REM ** End of mission.
4420 FOR I=1TO500:NEXT
4430 GOSUB4060:PRINT"Well done! ";
4440 PRINT"Mission accomplished."
4450 GOSUB4870
4460 IFR<=40THENPRINT"You made it - slowly!";
4470 IFR>40THENPRINT"Good work...";
4480 IFR>80THENPRINT"Fantastic...";
```

```
4490 IFR>100THENPRINT"Unbelievable!";
4500 PRINT
4510 PRINT:GOSUB4870
4520 PRINT"Captain "N$", on this mission"
4530 PRINT"you destroyed"K9-K"Klingons in";
4540 PRINTFNR(D9-D)"stardates,"
4550 PRINT" with"D"stardates left."
4560 IFK>0THENPRINTK"Klingons not destroyed."
4570 PRINT"Your rating is"R;
4580 PRINT"percent of standard."
4590 PRINTC9-C"out of"C9"crew died ";
4600 PRINT("FNR((C9-C)/C9*100);
4610 PRINT"percent).";
4620 IFB=B9THEN4660
4630 PRINTB9-B"starbase";
4640 IFB9-B>1THENPRINT"s were destroyed!"
4650 IFB9-B=1THENPRINT" was destroyed!"
4660 IFF8=0GOTO4690
4670 PRINTF8"photon torpedoes were fired, ";
4680 PRINTF7"missed."
4690 PRINT:Q$="":INPUT"Another mission ";Q$
4700 IFQ$<>"YES"ANDQ$<>"NO"GOTO4740
4710 IFQ$="YES"GOTO240
4720 PRINT"Farewell, Captain ";N$;".";
4730 GOTO5800
4740 PRINT"YES or NO only!":GOTO4690
4750 GOSUB4060:PRINT"Too late, the ";
4760 PRINT"Federation has been conquered!"
4770 GOTO4510
4780 GOSUB4060:PRINT"USS Enterprise ";
4790 PRINT"destroyed. No energy left."
4800 C=0:GOTO4510
4810 GOSUB4060:PRINT"All the crew are dead,";
4820 PRINT" including you.":GOTO4510
4830 GOSUB4060:PRINT"Mission abandoned."
4840 GOTO4510
4850 REM
4860 REM ** Rating.
4870 IFD9-D=0THENR=0:RETURN
4880 R=FNR((K9-K)/(D9-D)*100)
4890 RETURN
4900 REM
4910 REM ** List of commands.
4920 GOSUB4060
4930 PRINT"Enter one of these commands:"
4940 PRINT" H Help          S Short Range Scan"
4950 PRINT" W Warp Drive    L Long Range Scan"
4960 PRINT" P Phasor         T Photon Torpedo"
4970 PRINT" R Status Report"
4980 PRINT" D Damage Report"
4990 PRINT" G Galactic Record"
5000 PRINT" A Abandon Mission"
5010 PRINT:GOTO860
5020 REM
5030 REM ** Update time and damage.
5040 D2=D
```



```
5050 D=FNR(D-D1):IFD<0THEND=0
5060 IFINT(D+.5)=INT(D2+.5)THENRETURN
5070 FORI=0TO6:D(I)=D(I)-1:IFD(I)<0THEND(I)=0
5080 NEXT:RETURN
5090 REM
5100 REM ** Get name.
5110 N1$="":INPUT"What is your name ";N1$
5120 IFN1$=""THENPRINT"Please...";:GOTO5110
5130 FORI=1TOLEN(N1$)
5140 A=ASC(MID$(N1$,I,1))
5150 IFI9=1ORI=1THENIFA>=96THENA=A-32
5160 IFI9=1ORI=1GOTO5190
5170 IFA<=57GOTO5190
5180 IFA<=95THENA=A+32
5190 N$=N$+CHR$(A)
5200 I9=0:IFA=32THENN2$="":GOTO5220
5210 N2$=N2$+CHR$(A)
5220 IFA<=57THENI9=1
5230 IFN2$="Mc"THENI9=1
5240 IFN2$="Mac"ORN2$="Mc"THENI8=1
5250 NEXT
5260 IFI8=1THENPRINT"A scotsman. Eh!!"
5270 X=RND(-.01*ASC(N$)):RETURN
5280 REM
5290 REM Time input and return value.
5300 Q1$=""
5310 FORI9=1TO2000:REM 1000 FOR 2MHz clock
5320 I8=USR(0):IFI8>0GOTO5340
5330 NEXT:PRINT:GOTO770
5340 IFI8<>13GOTO5360
5350 PRINT:ONI7GOTO2970,1750,3940,880
5360 Q1$=Q1$+CHR$(I8)
5370 PRINTCHR$(I8);:GOTO5310
5380 REM
5390 REM ** Move Klingons.
5400 IFQ5=0THENRETURN
5410 Q3=0:FORI=0TO7:FORJ=0TO7
5420 IFINT(A(I,J)/100)=0GOTO5560
5430 IFRND(1)>Q5GOTO5560
5440 P=I:G=J
5450 P=P-SGN(P-U)
5460 G=G-SGN(G-V)
5470 IFP=IANDG=JGOTO5560
5480 A(I,J)=A(I,J)-100
5490 A(P,G)=A(P,G)+100
5500 IFP<>UORG<>VGOTO5560
5510 Q8=Q8+1:I$=K$:GOSUB1230
5520 FORM=0TO62:IFKP(M)=M9GOTO5540
5530 NEXT
5540 KP(M)=300:KX(M)=X1:KY(M)=Y1
5550 KP(M+1)=M9:M1=M1+1
5560 NEXT J:NEXT
5570 IFQ8=0GOTO5620
5580 PRINT"***"Q8"Klingon";
5590 IFQ8>1THENPRINT"s";
5600 PRINT" detected entering quadrant ***"
```

```
5610 G(U,V)=A(U,V)
5620 IFM1=0THENRETURN
5630 FORM=0TO62:IFKP(M)=M9THENRETURN
5640 IFKP(M)=0GOTO5680
5650 IFRND(1)>Q5/2GOTO5680
5660 S$(KX(M),KY(M))=X$:I$=K$:GOSUB1230
5670 KX(M)=X1:KY(M)=Y1
5680 NEXT
5690 REM
5700 REM ** Output symbols.
5710 E$=CHR$(185):K$=CHR$(191):B$=CHR$(186)
5720 X$="." :S$="*"
5730 REM Remove the 'REM' from lines 5750 & 5780
5740 REM to convert to teleprinter output.
5750 REM E$="E":K$="K":B$="B":X$="." :S$="*"
5760 L$=CHR$(129):L$=" "+L$+L$+L$
5770 G$=CHR$(129):G$=" "+G$+G$+G$
5780 REM L$=" ---":G$=" ---"
5790 RETURN
5800 END
Ok
```

```
10 REM          **** CUBIST ART ****
20 REM
30 REM By D. R. Hunt          December 1979
40 REM
50 REM For Nascom 1 or 2, using Nas-sys 1 and
60 REM          NAS-GRA V.3 Graphics.
70 REM
80 CLS:T$="CUBIST ART":FORI=1TOLEN(T$)
90 POKE3035+I,ASC(MID$(T$,I,1)):NEXT
100 A=RND(1)*95:B=RND(1)*43+1:C=RND(1)*95
110 D=RND(1)*43+1:A1=-1:A2=-1:IFA<CTHENA1=1
120 IFB<DTHENA2=1
130 FORI=ATOCSTEPSA1:FORJ=BTODSTEPSA2
140 A3=POINT(I,J):IFA3=1THENRESET(I,J):GOTO160
150 SET(I,J)
160 NEXT:NEXT:GOTO100
Ok
```

```
10 REM **** 3D NOUGHTS AND CROSSES ****
20 REM
30 REM Written for Nascom 1/2 fitted with
40 REM NAS-SYS 1 or 'T' series monitors
50 REM by Richard Beal June 1979.
60 REM
70 CLS:H$="3-D NOUGHTS AND CROSSES"
80 FOR I=1 TO LEN(H$)
90 POKE 3030+I,ASC(MID$(H$,I,1)):NEXT
100 N$="":INPUT"What is your name ";N$
110 IF N$=""GOTO100
120 R=RND(-.01*ASC(N$))
130 PRINT"Enter each move as a 3 digit number,"
140 PRINT"giving level, row and column, or"
150 PRINT"enter 888 to display board again, or"
160 PRINT"enter 999 to resign."
170 PRINT"You are 'X' and I am 'O'."
180 PRINT"Good luck, "N$"."
190 A=0:C=0:D=0:E=0:G=0:N=0:T=0
200 DIM M(63,6),N(75),E(18),D(63),W(63)
210 DIM P(63),H(7),C(10),G(63),Q(15),B(13)
220 F=-1
230 FOR A=0 TO 6:FOR B=0 TO 63:READ M(B,A)
240 NEXT:NEXT
250 FOR A=0 TO 15:READ Q(A):NEXT
260 FOR A=0 TO 7:READ H(A):NEXT
270 FOR A=0 TO 63:READ W(A):NEXT
280 FOR A=0 TO 10:READ C(A):NEXT
290 FOR A=0 TO 63:READ P(A):NEXT
300 FOR A=0 TO 13:READ B(A):NEXT
310 FOR A=0 TO 75:N(A)=0:NEXT
320 FOR A=0 TO 63:G(A)=0:NEXT
330 FOR A=0 TO 18:E(A)=0:NEXT
340 B$(0)=" .":B$(1)=" X":B$(2)=" O"
350 M=1:Q=76
360 IF F=-1 THEN GOSUB1010
370 A$=""
380 INPUT"Would you like me to go first ";A$
390 IF A$="YES" OR A$="NO"GOTO 410
400 PRINT"ERROR - Enter YES or NO":GOTO 370
410 Z=A$="YES"
420 IF Z GOTO 700
430 PRINT"Your move"M+Z"please ";
440 F=0:INPUT F
450 IF F<>999 GOTO 470
460 PRINT"You resigned.":GOTO 1160
470 IF F<>888 GOTO 490
480 F=-1:GOSUB1010:GOTO 430
490 IF F<>INT(F) OR F<111 OR F>444 GOTO 560
500 D=INT(F/100):IF D<1 OR D>4 GOTO 560
510 C=F-D*100:E=INT(C/10)
520 IF E<1 OR E>4 GOTO 560
530 C=C-E*10:IF C<1 OR C>4 GOTO 560
540 F=16*F-159*INT(F/10)-6*INT(F/100)-21
550 IF F>=0 AND F<=63 GOTO 570
560 PRINT"Impossible move.":GOTO 430
```

```
570 IF G(F)=0 GOTO 590
580 PRINT"Position occupied.":GOTO 430
590 G(F)=1
600 FOR B=0 TO 6+3*(M(F,6)=76)
610 N=N(M(F,B))
620 E=E(M(F,B)/4)
630 IF N<>3 GOTO 650
640 PRINT"OK - You win "N$!":GOTO 1160
650 IF N=INT(N/4)*4 AND N>3 THEN Q=Q-1
660 IF Q=0 GOTO 1150
670 N(M(F,B))=N+1
680 E(M(F,B)/4)=E-((E-INT(E/8)*8)<7)
690 NEXT
700 N=INT((M-1)/4)*3
710 P=-10000
720 FOR A=0 TO 63
730 IF G(A) GOTO 850
740 T=0:E=0
750 B=6+3*(M(A,6)=76)
760 G=(B/3-1)*7
770 FOR C=0 TO B
780 D=Q(N(M(A,C)))
790 T=T+W(N+D)+P(E(M(A,C)/4))*B(G+C)
800 E=E+H(D)
810 NEXT
820 IF E>10 THEN E=10
830 T=T+C(E)
840 IF T+RND(1)>P THEN P=T+.5:F=A:T2=E
850 NEXT
860 G(F)=2
870 GOSUB1010
880 FOR B=0 TO 6+3*(M(F,6)=76)
890 N=N(M(F,B))
900 E=E(M(F,B)/4)
910 IF N<>12 GOTO 930
920 PRINT"I win!!!":GOTO 1160
930 IF N<4 AND N<>0 THEN Q=Q-1
940 IF Q=0 GOTO 1150
950 N(M(F,B))=N+4
960 E(M(F,B)/4)=E-8*(E<56)
970 NEXT
980 M=M+1:GOTO 430
990 REM
1000 REM Display board
1010 PRINT"      1 2 3 4      1 2 3 4      1 2 3 4";
1020 PRINT"      1 2 3 4"
1030 FOR C=0 TO 3
1040 PRINT C+1" ";
1050 FOR E=0 TO 3
1060 IF E<>0 THEN PRINT"  ";
1070 R=E*4+C
1080 PRINT B$(G(R));B$(G(R+16));
1090 PRINT B$(G(R+32));B$(G(R+48));
1100 NEXT:PRINT:NEXT
1110 IF F=-1 THEN RETURN
1120 C=INT(F/16):E=F-16*C:D=INT(E/4):E=E-4*D
```

```
1130 PRINT"My move was"D*100+E*10+C+111
1140 RETURN
1150 PRINT"Definitely a draw."
1160 A$="":INPUT"Do yo want another game ";A$
1170 IF A$="YES" OR A$="NO"GOTO 1190
1180 PRINT"Error - Enter YES or NO":GOTO 1160
1190 IF A$="YES"GOTO 310
1200 PRINT"Goodbye "N$"." :GOTO 2120
1210 REM
1220 REM Line pointers M(63,6) (448)
1230 DATA 0, 0, 0, 0, 1,37,38, 1
1240 DATA 2,41,42, 2, 3, 3, 3, 3
1250 DATA 32, 4, 4,35, 5, 5, 5, 5
1260 DATA 6, 6, 6, 6,44, 7, 7,47
1270 DATA 32, 8, 8,35, 9, 9, 9, 9
1280 DATA 10,10,10,10,44,11,11,47
1290 DATA 12,12,12,12,13,37,38,13
1300 DATA 14,41,42,14,15,15,15,15
1310 DATA 16,33,34,28,16,20,24,28
1320 DATA 16,20,24,28,16,45,46,28
1330 DATA 17,21,25,29,36,21,25,39
1340 DATA 40,21,25,43,17,21,25,29
1350 DATA 18,22,26,30,36,22,26,39
1360 DATA 40,22,26,43,18,22,26,30
1370 DATA 19,33,34,31,19,23,27,31
1380 DATA 19,23,27,31,19,45,46,31
1390 DATA 32,20,24,35,36, 1, 1,39
1400 DATA 40, 2, 2,43,44,20,24,47
1410 DATA 4,33,34, 4,17,37,38,29
1420 DATA 17,41,42,29, 7,45,46, 7
1430 DATA 8,33,34, 8,18,37,38,30
1440 DATA 18,41,42,30,11,45,46,11
1450 DATA 32,23,27,35,36,13,13,39
1460 DATA 40,14,14,43,44,23,27,47
1470 DATA 48,65,66,52,57,48,52,61
1480 DATA 58,52,48,62,52,69,70,48
1490 DATA 49,56,60,53,64,49,53,67
1500 DATA 68,53,49,71,53,59,63,49
1510 DATA 50,60,56,54,68,50,54,71
1520 DATA 64,54,50,67,54,63,59,50
1530 DATA 51,69,70,55,61,51,55,57
1540 DATA 62,55,51,58,55,65,66,51
1550 DATA 56,76,76,60,76,76,76,76
1560 DATA 76,76,76,76,59,76,76,63
1570 DATA 76,76,76,76,76,57,61,76
1580 DATA 76,58,62,76,76,76,76,76
1590 DATA 76,76,76,76,76,61,57,76
1600 DATA 76,62,58,76,76,76,76,76
1610 DATA 60,76,76,56,76,76,76,76
1620 DATA 76,76,76,76,63,76,76,59
1630 DATA 64,76,76,67,76,76,76,76
1640 DATA 76,76,76,76,68,76,76,71
1650 DATA 76,76,76,76,76,65,66,76
1660 DATA 76,69,70,76,76,76,76,76
1670 DATA 76,76,76,76,76,69,70,76
1680 DATA 76,65,66,76,76,76,76,86
```

```

1690 DATA 68,76,76,71,76,76,76,76
1700 DATA 76,76,76,76,64,76,76,67
1710 DATA 72,76,76,73,76,76,76,76
1720 DATA 76,76,76,76,74,76,76,75
1730 DATA 76,76,76,76,76,72,73,76
1740 DATA 76,74,75,76,76,76,76,76
1750 DATA 76,76,76,76,76,75,74,76
1760 DATA 76,73,72,76,76,76,76,76
1770 DATA 75,76,76,74,76,76,76,76
1780 DATA 76,76,76,76,73,76,76,72
1790 REM
1800 REM Line pointers Q(15)
1810 DATA 0,4,5,6,1,7,7,7,2,7,7,7,3,7,7,7
1820 REM
1830 REM Rows of two counters H(7)
1840 DATA 0,0,5,0,0,1,0,0
1850 REM
1860 REM Main priorities W(63)
1870 DATA 5,10, 1,4000, 1, 4,700,0
1880 DATA 5,10, 1,4000, 1, 5,700,0
1890 DATA 5,15, 5,4000, 5, 6,700,0
1900 DATA 5,15,25,4000, 1,12,700,0
1910 DATA 25,20,40,4000, 1,15,700,0
1920 DATA 25,30,40,4000, 1,15,700,0
1930 DATA 25, 1, 1,4000,10,40,700,0
1940 DATA 25, 1, 1,4000,10,40,700,0
1950 REM
1960 REM Rows of two priorities C(10)
1970 DATA 0,0,160,180,200,0,20,180,200,200,400
1980 REM
1990 REM Planes priorities P(63)
2000 DATA 0, 0, 0,5,50,100,200,300
2010 DATA 0, 0, 0,0, 2, 60, 80,100
2020 DATA 0, 0, 0,0, 0, 1, 40, 60
2030 DATA 5, 0, 0,0, 0, 0, 1, 20
2040 DATA 55, 5, 0,0,-5, 0, 0, 0
2050 DATA 60, 50,10,0, 0,-20, 0, 0
2060 DATA 120,100,20,0, 0, 0,-20, 0
2070 DATA 150,120,25,0, 0, 0, 0,-20
2080 REM
2090 REM Diagonal multipliers B(13)
2100 DATA 1, 1,1.5, 1, 1, 1, 1
2110 DATA 1.5,1.5,1.5,1.5,1.5,1.5,1.5
2120 END
Ok

```

Non Existant Lines.
=====

Did you know that it is perfectly legal to have a line 0 in the 8K BASIC ? Very useful for inserting a GOTO if your program starts at line 1, you don't have a BASIC renumber package, and you've left out something vital. Try it.

```
10 REM          *** CALENDER ***
20 REM
30 REM   By D. R. Hunt.          2 October 1979
40 REM
50 REM   Suitable for Nascom 1/2 fitted with
60 REM   NASBUG 'T' series or NAS-SYS monitors.
70 REM   This program occupies approx. 1.5K.
80 REM
90 REM ** Set up machine code INKEY$ command
100 IF PEEK(1)<>0 THEN RESTORE 470
110 DOKE 4100,3328:FOR A=3328 TO 3342 STEP 2
120 READ B:DOKE A,B:NEXT:CLS
130 REM ** Get inputs
140 PRINT "Calender for the years 1601 to 2399"
150 INPUT "Do you require continuous output ";I$
160 F1=0:IF LEFT$(I$,1)="Y" THEN F1=1
170 INPUT "What year ";X
180 IF X>=1601 AND X<=2399 THEN 210
190 PRINT "Out of range !":GOTO 170
200 REM ** Calculate starting day
210 C=5:FOR J=1600 TO X
220 F=0:IF J/4<>INT(J/4) THEN 250
230 IF (J-1700)*(J-1800)*(J-1900)*(J-2100)*(J-2200)*(J-2300)=0 THEN 250
240 F=1:C=C+1
250 C=C+1:IF C>=7 THEN C=C-7
260 NEXT:RESTORE 500:IF F=1 THEN C=C-1
270 REM ** Print calender
280 CLS:FOR R=1 TO 12:READ A$
290 PRINT TAB(20-INT((LEN(A$)+6)/2));A$;
300 PRINT " ";X:READ B:IF F<>1 THEN 320
310 IF A$="FEBRUARY" THEN B=B+1
320 PRINT"=====
330 PRINT"!SUN  MON   TUE    WED    THU    FRI    SAT!"
340 PRINT"=====
350 FOR D=1 TO B:PRINT TAB(6*C+1);
360 IF D<10 THEN PRINT" ";
370 PRINT D;:C=C+1:IF C=7 THEN PRINT:C=0
380 NEXT:IF C<>0 THEN PRINT
390 PRINT"=====
400 PRINT:PRINT
410 IF F1=1 OR R=12 THEN NEXT:GOTO150
420 PRINT"Hit any key to continue."
430 A=USR(0):IF A<0 THEN 430
440 PRINT:NEXT:GOTO 150
450 REM ** Data for machine code INKEY$
460 DATA 25055,1080,-53,536,-20665,3370,-5664,0
470 DATA 27085,14336,-13564,6399,18178,10927
480 DATA -8179,233
490 REM ** Data for months and days
500 DATA "JANUARY",31,"FEBRUARY",28
510 DATA "MARCH",31,"APRIL",30,"MAY",31
520 DATA "JUNE",30,"JULY",31,"AUGUST",31
530 DATA "SEPTEMBER",30,"OCTOBER",31
540 DATA "NOVEMBER",30,"DECEMBER",31
Ok
```

```
10 REM          *** MAGIC LABYRINTH ***
20 REM
30 REM A version of Kingly Orb, based on the
40 REM program published in Practical Computing,
50 REM adapted for 16K Nascom 1/2 using Nasbug
60 REM or Nas-sys monitors, by Howard Birkett.
70 REM With minor alterations by D. R. Hunt.
80 REM
90 REM December 1979
100 REM
110 CLEAR 1000:DIM A(100,3),D(5)
120 RESTORE 2150:IF PEEK(1)<>0 THEN RESTORE 2170
130 DOKE 4100,3328:FOR K=3328 TO 3342 STEP 2
140 READ J:DOKE K,J:NEXT:RESTORE
150 DEF FNR(Z)=INT(RND(1)*Z+1)
160 CLS:GOSUB 2530:PRINT:PRINT
170 I$="":INPUT"Do you want instructions ";I$
180 IF I$="NO"THEN 370
190 IF I$<>"YES"THEN PRINT"Come on ..":GOTO 170
200 CLS:GOSUB 2530
210 PRINT"You are in a five level labyrinth, each level"
220 PRINT"is divided into a grid of 100 x 100 squares."
230 PRINT"There are walls around the labyrinth, and"
240 PRINT"you will be penalised if you bump into one. You"
250 PRINT"must remember where you are. Your quest is to"
260 PRINT"find the 5 Objects of State, one of each in"
270 PRINT"each level. During your quest, you will find 3"
280 PRINT"kinds of creature, good, evil, and neutral. The"
290 PRINT"good will help you, the evil will cast spells"
300 PRINT"to destroy you, and neutral will help or hinder"
310 PRINT"as they feel fit. You start with 3 counter"
320 PRINT"spells. To see the list of spells type -1; to"
330 PRINT"run away, type 0. Moves are made by typing the"
340 PRINT"number of grid squares, north and east of the"
350 PRINT"origin required. The compass points + & - only.";
360 K=USR(0):IF K<0 THEN 360
370 CLS:GOSUB 2530:GOSUB 2550:L=1:GOSUB 2570
380 PRINT:PRINT
390 PRINT"Good luck your Imperial Highness !"
400 PRINT" ( you'll need it ).":PRINT:PRINT
410 P=3:S=400:C0=0:D=999:P1=9:P2=9
420 FOR I=1 TO 100:A(I,1)=FNR(10)
430 A(I,2)=FNR(3):NEXT
440 X1=FNR(100):X2=FNR(100)
450 D1=FNR(100):D2=FNR(100)
460 IF D1=X1 OR D2=X2 GOTO 450
470 IF FNR(1000)>600 GOTO 490
480 I=FNR(100):GOTO 760
490 REM
500 REM ** Print outputs
510 IF C0<1 THEN 550
520 X8=D1:X9=D2:GOSUB 2420:D=Z
530 IF C0<2 GOTO 550
540 P1=SGN(D1-X1):P2=SGN(D2-X2)
550 PRINT"You're at";X1;X2;
560 IF C0<1 GOTO 640
```



```
570 PRINT ", and";INT(D);"from Magic Spot"
580 IF C0<2 GOTO 640
590 P1$="-":IF P1=-1 THEN P1$="+"
600 IF P1=0 THEN P1$="0"
610 P2$="-":IF P2=-1 THEN P2$="+"
620 IF P2=0 THEN P2$="0"
630 PRINT"Your Compass reads N ";P1$;" E ";P2$
640 IF D=0 GOTO 1260
650 IF C0>0 GOTO 670
660 REM
670 REM ** Get direction input
680 PRINT: INPUT "Direction ";C1,C2
690 C1=INT(C1):C2=INT(C2)
700 X8=X1+C1:X9=X2+C2:GOSUB 2420
710 IF Z>20 GOTO 1550
720 X1=X1+C1
730 X2=X2+C2
740 GOSUB 1680:GOTO 470
750 REM
760 REM ** Find hazards
770 READ C$(1),C$(2),C$(3)
780 FOR J=1 TO A(1,1)
790 READ B$,S0,D(0),D(1),D(2),D(3),D(4):NEXT
800 RESTORE
810 S0=S0*L:T=A(1,2)
820 IF S<=30*L OR P<>10 GOTO 840
830 T=2
840 PRINT"You've found a";C$(T);" ";B$;
850 PRINT" power";S0
860 ON T GOTO 890,960,870
870 IF FNR(10)<5 AND (S<=30*L OR P<10) GOTO 890
880 GOTO 1110
890 C0=C0+1:IF C0>3 GOTO 910
900 ON C0 GOTO 930, 940, 950
910 IF S<L*30 GOTO 950
920 P=P+1:PRINT"You have been taught Spell No.";P:GOTO 490
930 PRINT"You have been given a Distance Meter":GOTO 490
940 PRINT"You have been given a Compass":GOTO 490
950 PRINT"Your strength has been increased":S=S+50*L:GOTO 490
960 D3=S-S0:IF D3<0 GOTO 1110
970 REM
980 REM ** Get spells
990 PRINT: INPUT "What Spell No. ";V
1000 IF V<>-1 GOTO 1020
1010 PRINT:GOSUB 2210 :GOTO 990
1020 IF V=0 OR (V>10 AND V<10+L AND L>1) GOTO 1400
1030 IF V>P OR V<0 THEN 1090
1040 FOR I2=0 TO 4
1050 IF D(I2)=V THEN I2=4:NEXT I2:GOTO 1080
1060 NEXT I2
1070 IF FNR(5)<>3 GOTO 1250
1080 GOTO 1100
1090 PRINT"You cannot use Spell No.";V:GOTO 970
1100 S0=S0-V*10:IF S0<=0 GOTO 1140
1110 P3=S0/10/L:V=FNR(INT(P3))
1120 PRINT"The ";B$;" uses Spell No.";V;"on you"
```

```
1130 PRINT TAB(4)"Its strength is "S0
1140 IF S0<=0 GOTO 1210
1150 S=INT(S-V*10):PRINT TAB(4)"Your strength is";S
1160 IF S>0 GOTO 970
1170 PRINT"You have failed in your Quest.";
1180 PRINT" You were"
1190 PRINT"fouly murdered by a ";B$;". Bad luck."
1200 GOTO 2470
1210 PRINT"The "B$" is no more."
1220 PRINT"The powers of Good win through again."
1230 PRINT"Now continue with your Quest."
1240 PRINT:GOTO 490
1250 PRINT"Your spell had no affect":GOTO 1110
1260 PRINT"You are on the Magic Spot"
1270 RESTORE 2080:FOR I=1 TO L:READ A$:NEXT
1280 RESTORE:PRINT A$:L=L+1
1290 IF L=6 GOTO 1390
1300 IF P=10 GOTO 1360
1310 PRINT"You have met the Good Guide to the next level."
1320 PRINT"Extra Wisdom means that you learn";
1330 P=P+1:PRINT" Spell No."P
1340 S=S+INT(S*(RND(1)*.5))
1350 PRINT"You have been given more strength."
1360 PRINT"You have been taken to another Level"
1370 GOSUB 2570
1380 GOTO 440
1390 GOTO 2470
1400 IF V>P GOTO 1730
1410 REM
1420 REM ** Get direction if pursued
1430 INPUT "Direction to run in ";A,B
1440 A=INT(A):B=INT(B)
1450 X8=X1+A:X9=X2+B:GOSUB 2420
1460 IF Z<=20 GOTO 1490
1470 PRINT"You cannot go that far"
1480 GOTO 1410
1490 X1=X1+A:X2=X2+B:GOSUB 1680
1500 IF FNR(1000)>600 GOTO 1530
1510 PRINT"The "B$" has not followed you"
1520 GOTO 470
1530 PRINT"The "B$" is in hot pursuit"
1540 GOTO 1110
1550 PRINT"You cannot go that far"
1560 GOTO 520
1570 X1=1
1580 GOTO 1640
1590 X1=100
1600 GOTO 1640
1610 X2=1
1620 GOTO 1640
1630 X2=100
1640 S=S*.9
1650 PRINT"You have bumped into a wall and lost"
1660 PRINT"10% of your strength"
1670 RETURN
1680 IF X1<1 GOTO 1570
```

```
1690 IF X1>100 GOTO 1590
1700 IF X2<1 GOTO 1610
1710 IF X2>100 GOTO 1630
1720 RETURN
1730 S(V-10)=S(V-10)+1
1740 IF S(V-10)=1 GOTO 1780
1750 PRINT "You have already used that Object ";
1750 PRINT "of State"
1770 GOTO 970
1780 IF V=11 AND B$="Rock Monster" GOTO 1840
1790 IF V=12 AND B$="Mummy" GOTO 1840
1800 IF V=13 AND B$="Giant" GOTO 1840
1810 IF V=14 AND B$="Wizard" GOTO 1840
1820 PRINT "The "B$" is undamaged"
1830 GOTO 1110
1840 IF FNR(1000)<100 GOTO 1820
1850 PRINT "The Object of State won"
1850 GOTO 1230
1870 DATA " Good","n Evil"," Neutral"
1880 DATA Dragon,57
1890 DATA 4,6,7,9,10
1900 DATA Witch,25
1910 DATA 2,3,5,8,10
1920 DATA Wizard,82
1930 DATA 2,3,5,8,10
1940 DATA Vampire,12
1950 DATA 4,6,7,9,10
1960 DATA Rock Monster,36
1970 DATA 4,7,8,9,10
1980 DATA Mummy,47
1990 DATA 3,6,8,10,2
2000 DATA Gold Horse,78
2010 DATA 3,4,6,10,8
2020 DATA Sand Man,34
2030 DATA 1,4,10,7,5
2040 DATA Giant,75
2050 DATA 10,3,6,8,7
2060 DATA Water Worm,30
2070 DATA 10,8,3,2,1
2080 DATA You have found the Great Ring
2090 DATA You have found the Kingly Orb
2100 DATA The Robes of State are yours
2110 DATA The Sword of Peace is yours
2120 DATA You are now the true Monarch of Oz
2130 REM
2140 REM ** Data for machine code input
2150 DATA 25055,1080,-53,536,-20665
2160 DATA 3370,-5664,0
2170 DATA 27085,14336,-13564,6399,18178
2180 DATA 10927,-8179,233
2190 REM
2200 REM ** List of spells etc
2210 PRINT "The Spells are ...."
2220 PRINT TAB(5);
2230 PRINT "(1) Landslide      (6) Make swamp"
2240 PRINT TAB(5);
```

```

2250 PRINT"(2) Wind          (7) Rain of Holy Water"
2260 PRINTTAB(5);
2270 PRINT"(3) Fireball      (8) Earthquake"
2280 PRINTTAB(5);
2290 PRINT"(4) Make Pool     (9) Rainstorm"
2300 PRINTTAB(5);
2310 PRINT"(5) Create Hole   (10)Thunderstorm"
2320 PRINT:PRINT"The Objects of State destroy ...."
2330 PRINTTAB(5);
2340 PRINT"(11) Great Ring    ==  Rock Monster
2350 PRINTTAB(5);
2360 PRINT"(12) Kingly Orb    ==  Mummy"
2370 PRINTTAB(5);
2380 PRINT"(13) Robes of State ==  Giant"
2390 PRINTTAB(5);
2400 PRINT"(14) Sword of Peace ==  Wizard"
2410 RETURN
2420 REM
2430 REM ** Make move
2440 Z=SQR((X1-X8).2+(X2-X9).2):RETURN
2450 REM
2460 REM ** End
2470 PRINT:INPUT "Another game ";A$
2480 IF LEFT$(A$,1)="Y" THEN370
2490 PRINT"Goodbye"
2500 PRINT:END
2510 REM
2520 REM ** Put up titles
2530 T$="MAGIC LABYRINTH":FOR K=1 TO LEN(T$)
2540 POKE 3019+K,ASC(MID$(T$,K,1)):NEXT K:RETURN
2550 T$="You are on level .":FOR K=1 TO LEN(T$)
2560 POKE 3043+K,ASC(MID$(T$,K,1)):NEXT K:RETURN
2570 REM
2580 REM ** Put in current level
2590 POKE 3061,48+L:RETURN
Ok

```

LITTLE KNOWN FACTS THAT NO-ONE SEEMS TO CARE ABOUT

=====

Did you know that line number arguments can be appended to the 'RESTORE' command in the 8K Basic. Thus:

```

20 DATA 20,20,37,196,20,53,30
30 DATA 20,21,44,44,196,37,77,60

```

.
etc

```

100 RESTORE 30: IF X=Y THEN RESTORE 20
etc

```

Note that in the above example the 'RESTORE' command restores to the second line of 'DATA', not the first as would be more usual. Further, the 'RESTORE' to the first line is made conditional upon X and Y. Interesting ain't it!

```
10 REM          *** ELIZA ***
20 REM
30 REM From the program by the same name by
40 REM Jeff Shrager, published by Creative
50 REM Computing. Adapted for Nascom 1/2 by
60 REM D. R. Hunt          October 1979
70 REM
80 REM Requires approx. 16K RAM space.
90 REM Suitable for NASBUG T4 and NAS-SYS.
100 REM
110 REM ** Initialization
120 REM
130 CLEAR 1500: DIM S(36), R(36), N(36)
140 REM
150 REM ** Set up machine code input.
160 REM
170 RESTORE 2740: X1=31: X2=29
180 IF PEEK(1)=0 THEN RESTORE 2710: X1=13: X2=8
190 DOKE 4100, 3328: FOR I9=3328 TO 3342 STEP 2
200 READ I8: DOKE I9, I8: NEXT
210 REM
220 REM ** Change A for use with a printer
230 REM
240 A=47
250 WIDTH A: N1=36: N2=14: N3=112
260 RESTORE 2630
270 FOR X=1 TO N1: READ S(X), L: R(X)=S(X)
280 N(X)=S(X)+L-1: NEXT
290 CLS: PRINT "Hello, I'm a Nascom specially trained in"
300 PRINT "psychoanalysis. Please tell me your problems."
310 REM
320 REM ** User input section.
330 REM
340 PRINT: I$=" ": B1=0
350 B=USR(0): IF B<0 THEN 350
360 IF B=X2 AND I$=" " THEN 350
370 IF B1=175 AND B<>X2 AND B<>X1 THEN 350
380 PRINT CHR$(B);: IF B=X1 THEN 430
390 IF B=X2 THEN 420
400 IF B>92 THEN B=B-32
410 I$=I$+CHR$(B): B1=B1+1: GOTO 350
420 I$=LEFT$(I$, LEN(I$)-1): B1=B1-1: GOTO 350
430 I$=I$+" ": IF PEEK(1)<>0 THEN PRINT
440 REM
450 REM Strip out surplus punctuation.
460 REM
470 FOR L=1 TO LEN(I$)
480 IF MID$(I$, L, 1)="'" THEN I$=LEFT$(I$, L-1)+RIGHT$(I$, LEN(I$)-L): GOTO 480
490 IF MID$(I$, L, 1)="," THEN I$=LEFT$(I$, L-1)+RIGHT$(I$, LEN(I$)-L): GOTO 490
500 IF MID$(I$, L, 1)="." THEN I$=LEFT$(I$, L-1)+RIGHT$(I$, LEN(I$)-L): GOTO 500
510 IF MID$(I$, L, 1)="?" THEN I$=LEFT$(I$, L-1)+RIGHT$(I$, LEN(I$)-L): GOTO 510
520 IF L+4<=LEN(I$) THEN IF MID$(I$, L, 5)="SHUT " THEN PRINT "Shut up ....": :END
530 NEXT
540 IF I$=P$ THEN PRINT "Please don't repeat yourself!": GOTO 310
550 REM
560 REM ** Find key word in I$
```

```
570 REM
580 RESTORE
590 S=0:FOR K=1 TO N1
600 READ K$:IF S>0 THEN 640
610 FOR L=1 TO LEN(I$)-LEN(K$)+1
620 IF MID$(I$,L,LEN(K$))=K$ THEN S=K:T=L:F$=K$
630 NEXT L
640 NEXT K
650 IF S>0 THEN K=S:L=T:GOTO 680
660 K=36:GOTO 910: REM No keywords
670 REM
680 REM ** Take right part of string and
690 REM     conjugate it, using list of strings
700 REM     to be swapped.
710 REM
720 RESTORE 1410
730 C$=" "+RIGHT$(I$,LEN(I$)-LEN(F$)-L+1)+" "
740 FOR X=1 TO N2/2:READ S$,R$
750 FOR L=1 TO LEN(C$)
760 IF L+LEN(S$)>LEN(C$) THEN 800
770 IF MID$(C$,L,LEN(S$))<>S$ THEN 800
780 C$=LEFT$(C$,L-1)+R$+RIGHT$(C$,LEN(C$)-L-LEN(S$)+1)
790 L=L+LEN(R$):GOTO 830
800 IF L+LEN(R$)>LEN(C$) THEN 830
810 IF MID$(C$,L,LEN(R$))<>R$ THEN 830
820 C$=LEFT$(C$,L-1)+S$+RIGHT$(C$,LEN(C$)-L-LEN(R$)+1)
830 NEXT L
840 NEXT X
850 IF MID$(C$,2,1)=" "THEN C$=RIGHT$(C$,LEN(C$)-1)
860 FOR L=1 TO LEN(C$)
870 IF MID$(C$,L,1)<>"!" THEN 890
880 C$=LEFT$(C$,L-1)+RIGHT$(C$,LEN(C$)-L):GOTO 870
890 NEXT L
900 REM
910 REM ** Now using keyword number (K),
920 REM     get the reply.
930 REM
940 RESTORE 1480
950 FOR X=1 TO R(K):READ F$:NEXT
960 R(K)=R(K)+1:IF R(K)>N(K) THEN R(K)=S(K)
970 REM
980 REM ** Convert output string to lower case.
990 REM
1000 IF RIGHT$(F$,1)<>"*" THEN Z$=F$:GOTO 1020
1010 Z$=LEFT$(F$,LEN(F$)-1)+C$
1020 Z1$="":FOR L=1 TO LEN(Z$)
1030 Z=ASC(MID$(Z$,L,1)):IF L=1 THEN 1070
1040 IF MID$(Z$,L,3)<>" I " THEN 1060
1050 Z1$=Z1$+" I ":L=L+2:GOTO 1080
1060 IF Z>=65 THEN Z=Z+32
1070 Z1$=Z1$+CHR$(Z)
1080 NEXT L
1090 Z2$=RIGHT$(Z$,1)
1100 IF Z2$="."OR Z2$="?"OR Z2$="!"THEN 1170
1110 FOR L=LEN(Z1$) TO 1 STEP -1
1120 IF MID$(Z1$,L,1)=" " THEN NEXT
```

```
1130 Z1$=LEFT$(Z1$,L)+" ?"
1140 REM
1150 REM ** Justify to width held in A
1160 REM
1170 IF LEN(Z1$)<A THEN PRINT Z1$:GOTO 1230
1180 Z$=LEFT$(Z1$,A):FOR J=LEN(Z$) TO 1 STEP-1
1190 IF MID$(Z$,J,1)=" " THEN 1210
1200 NEXT
1210 Z$=LEFT$(Z$,J):PRINT Z$
1220 Z1$=RIGHT$(Z1$,LEN(Z1$)-J):GOTO 1170
1230 P$=I$:GOTO 310
1240 REM
1250 REM ** Program data follows
1260 REM
1270 REM ** Keywords
1280 REM
1290 DATA "CAN YOU","CAN I","YOU ARE","YOU'RE"
1300 DATA "I DON'T","I FEEL","WHY DON'T YOU"
1310 DATA "WHY CAN'T I","ARE YOU","I CAN'T"
1320 DATA "I AM","I'M","YOU ","I WANT","WHAT"
1330 DATA "HOW","WHO","WHERE","WHEN","WHY"
1340 DATA "NAME","CAUSE","SORRY","DREAM"
1350 DATA "HELLO","HI ","MAYBE"," NO","YOUR"
1360 DATA "ALWAYS","THINK","ALIKE","YES"
1370 DATA "FRIEND","COMPUTER","NOKEYFOUND"
1380 REM
1390 REM ** String data for conjugations
1400 REM
1410 DATA " ARE "," AM "," WERE "," WAS "
1420 DATA " YOU "," I "," YOUR "," MY "," IVE "
1430 DATA " YOUVE "," IM "," YOU'RE "," ME "
1440 DATA " !YOU "
1450 REM
1460 REM ** Replies
1470 REM
1480 DATA "DON'T YOU BELIEVE THAT I CAN*"
1490 DATA "PERHAPS YOU WOULD LIKE TO BE ABLE TO*"
1500 DATA "YOU WANT ME TO BE ABLE TO*"
1510 DATA "PERHAPS YOU DON'T WANT TO*"
1520 DATA "DO YOU WANT TO BE ABLE TO*"
1530 DATA "WHAT MAKES YOU THINK I AM*"
1540 DATA "DOES IT PLEASE YOU TO THINK I AM*"
1550 DATA "PERHAPS YOU WOULD LIKE TO BE"
1560 DATA "DO YOU SOMETIMES WISH YOU WERE*"
1570 DATA "DON'T YOU REALLY*"
1580 DATA "WHY DON'T YOU*"
1590 DATA "DO YOU WISH TO BE ABLE TO*"
1600 DATA "DOES THAT TROUBLE YOU ?"
1610 DATA "TELL ME MORE ABOUT SUCH FEELINGS."
1620 DATA "DO YOU OFTEN FEEL*"
1630 DATA "DO YOU ENJOY FEELING*"
1640 DATA "DO YOU REALLY BELIEVE I DON'T*"
1650 DATA "PERHAPS IN GOOD TIME I WILL*"
1660 DATA "DO YOU WANT ME TO*"
1670 DATA "DO YOU THINK YOU SHOULD BE ABLE TO*"
1680 DATA "WHY CAN'T YOU*"
```

1690 DATA"WHY ARE YOU INTERESTED IN WHETHER OR NOT I AM*"
1700 DATA"WOULD YOU PREFER IF I WERE NOT*"
1710 DATA"PERHAPS IN YOUR FANTASIES I AM*"
1720 DATA"HOW DO YOU KNOW YOU CAN'T*"
1730 DATA"HAVE YOU TRIED ?"
1740 DATA"PERHAPS YOU CAN NOW*"
1750 DATA"DID YOU COME TO ME BECAUSE YOU ARE*"
1760 DATA"HOW LONG HAVE YOU BEEN*"
1770 DATA"DO YOU BELIEVE IT IS NORMAL TO BE*"
1780 DATA"DO YOU ENJOY BEING*"
1790 DATA"WE WERE DISCUSSING YOU .. NOT ME."
1800 DATA"OH, I*"
1810 DATA"YOU'RE NOT REALLY TALKING ABOUT ME, ARE YOU ?"
1820 DATA"WHAT WOULD IT MEAN TO YOU IF I GOT*"
1830 DATA"WHY DO YOU WANT*"
1840 DATA"SUPPOSE YOU SOON GOT*"
1850 DATA"WHAT IF YOU NEVER GOT*"
1860 DATA"I SOMETIMES ALSO WANT*"
1870 DATA"WHY DO YOU ASK ?"
1880 DATA"DOES THAT QUESTION INTEREST YOU ?"
1890 DATA"WHAT ANSWER WOULD PLEASE YOU THE MOST ?"
1900 DATA"WHAT DO YOU THINK ?"
1910 DATA"ARE SUCH QUESTIONS ON YOUR MIND OFTEN ?"
1920 DATA"WHAT IS IT THAT YOU REALLY WANT TO KNOW ?"
1930 DATA"HAVE YOU ASKED ANYONE ELSE ?"
1940 DATA"HAVE YOU ASKED SUCH QUESTIONS BEFORE ?"
1950 DATA"WHAT ELSE COMES TO MIND WHEN YOU ASK THAT ?"
1960 DATA"NAMES DON'T INTEREST ME."
1970 DATA"I DON'T CARE ABOUT NAMES .. PLEASE GO ON."
1980 DATA"IS THAT THE REAL REASON ?"
1990 DATA"DON'T ANY OTHER REASONS COME TO MIND ?"
2000 DATA"DOES THAT REASON EXPLAIN ANTHING ELSE ?"
2010 DATA"WHAT OTHER REASONS MIGHT THERE BE ?"
2020 DATA"PLEASE DON'T APOLOGISE !"
2030 DATA"APOLOGIES ARE NOT NECESSARY."
2040 DATA"WHAT FEELINGS DO YOU HAVE WHEN YOU APOLOGISE ?"
2050 DATA"DON'T BE SO SENSITIVE !"
2060 DATA"WHAT DOES THAT DREAM SUGGEST TO YOU ?"
2070 DATA"DO YOU DREAM OFTEN ?"
2080 DATA"WHAT PERSONS APPEAR IN YOUR DREAMS ?"
2090 DATA"ARE YOU DISTURBED BY YOUR DREAMS ?"
2100 DATA"HOW DO YOU DO .. PLEASE STATE YOUR PROBLEM."
2110 DATA"YOU DON'T SEEM QUITE CERTAIN."
2120 DATA"WHY THE UNCERTAIN TONE ?"
2130 DATA"CAN'T YOU BE MORE POSITIVE ?"
2140 DATA"YOU AREN'T SURE ?"
2150 DATA"DON'T YOU KNOW ?"
2160 DATA"ARE YOU SAYING NO JUST TO BE NEGATIVE ?"
2170 DATA"YOU ARE BEING A BIT NEGATIVE."
2180 DATA"WHY NOT ?"
2190 DATA"ARE YOU SURE ?"
2200 DATA"WHY NO ?"
2210 DATA"WHY ARE YOU CONCERNED ABOUT MY*"
2220 DATA"WHAT ABOUT YOUR*"
2230 DATA"CAN YOU THINK OF A SPECIFIC EXAMPLE ?"
2240 DATA"WHEN ?"

2250 DATA"WHAT ARE YOU THINKING OF ?"
2260 DATA"REALLY, ALWAYS ?"
2270 DATA"DO YOU REALLY THINK SO ?"
2280 DATA"BUT YOU ARE NOT SURE YOU*"
2290 DATA"DO YOU DOUBT YOU*"
2300 DATA"IN WHAT WAY ?"
2310 DATA"WHAT RESEMBLANCE DO YOU SEE ?"
2320 DATA"WHAT DOES THE SIMILARITY SUGGEST TO YOU ?"
2330 DATA"WHAT OTHER CONNECTIONS DO YOU SEE ?"
2340 DATA"COULD THERE REALLY BE SOME CONNECTION ?"
2350 DATA"HOW ?"
2360 DATA"YOU SEEM QUITE POSITIVE."
2370 DATA"ARE YOU SURE ?"
2380 DATA"I SEE, TELL ME MORE."
2390 DATA"I UNDERSTAND."
2400 DATA"WHY DO YOU BRING UP THE TOPIC OF FRIENDS ?"
2410 DATA"DO YOUR FRIENDS WORRY YOU ?"
2420 DATA"DO YOUR FRIEND PICK ON YOU ?"
2430 DATA"ARE YOU SURE YOU HAVE ANY FRIENDS ?"
2440 DATA"DO YOU IMPOSE ON YOUR FRIENDS ?"
2450 DATA"PERHAPS YOUR LOVE FOR YOUR FRIENDS WORRIES YOU ?"
2460 DATA"DO COMPUTERS WORRY YOU ?"
2470 DATA"ARE YOU TALKING ABOUT ME IN PARTICULAR ?"
2480 DATA"ARE YOU FRIGHTENED BY MACHINES ?"
2490 DATA"WHY DO YOU MENTION COMPUTERS ?"
2500 DATA"WHAT DO YOU THINK MACHINES HAVE TO DO WITH YOUR PROBLEM ?"
2510 DATA"DON'T YOU THINK COMPUTERS CAN HELP PEOPLE ?"
2520 DATA"WHAT IS IT ABOUT MACHINES THAT WORRIES YOU ?"
2530 DATA"DON'T YOU HAVE ANY INTERESTING PSYCHOLOGICAL PROBLEMS ?"
2540 DATA"WHAT DOES THAT SUGGEST TO YOU ?"
2550 DATA"I SEE, IS THAT ALL YOU HAVE TO SAY ?"
2560 DATA"I'M NOT SURE I UNDERSTAND YOU FULLY."
2570 DATA"COME COME, ELUCIDATE YOUR THOUGHTS."
2580 DATA"CAN YOU ELABORATE ON THAT ?"
2590 DATA"THAT IS QUITE INTERESTING."
2600 REM
2610 REM ** Data for finding replies
2620 REM
2630 DATA 1,3,4,2,6,4,6,4,10,4,14,3,17,3,20,2
2640 DATA 22,3,25,3,28,4,28,4,32,3,35,5,40,9
2650 DATA 40,9,40,9,40,9,40,9,40,9,40,2,51,4
2660 DATA 55,4,59,4,63,1,63,1,64,5,69,5,74,2
2670 DATA 76,4,80,3,83,7,90,3,93,6,99,7,106,6
2680 REM
2690 REM ** Data for machine code input
2700 REM
2710 DATA 31711,1080,-53,536,-20665,3370
2720 DATA -5664,0
2730 REM
2740 DATA 27085,14336,-13564,6399,18178
2750 DATA 10927,-8179,233
OK

```
10 REM          **** CAMEL ****
20 REM
30 REM From the game of the same name by the
40 REM Heath Users Group, published by
50 REM Creative Computing.
60 REM
70 REM Adapted for Nascom 1/2 fitted with
80 REM NASBUG 'T' series or NAS-SYS monitors
90 REM by D. R. Hunt.      30 September 1979
100 REM
110 DATA 25055,1080,-53,536,-20665,3370,-5664,0
120 DATA 27085,14336,-13564,6399,18178,10927
130 DATA -8179,233
140 IF PEEK(1)<>0 THEN RESTORE 120
150 DOKE 4100,3328:FOR A=3328 TO 3342 STEP 2
160 READ B:DOKE A,B:NEXT
170 GOSUB 2130:SCREEN 8,7
180 INPUT "Would you like instructions ";D$
190 IF LEFT$(D$,1)<>"N" THEN 210
200 GOSUB 2130:SCREEN 1,7:GOTO 380
210 GOSUB 2130:PRINT TAB(16);"Welcome to Camel"
220 PRINT"You have stolen the priceless ";
230 PRINT"Idol belonging"
240 PRINT"to a tribe of knock-kneed pigmies. ";
250 PRINT"They are in"
260 PRINT"hot pursuit. From time to time ";
270 PRINT"you will be"
280 PRINT"asked for a command. ":GOSUB 2130
290 PRINT"You have 1 litre of water, which ";
300 PRINT"will last you"
310 PRINT"six drinks. You may replenish your ";
320 PRINT"water at an"
330 PRINT"oasis. If help finds you, you get ";
340 PRINT"half a litre"
350 PRINT"of water. If help does not ";
360 PRINT"find you after"
370 PRINT"command 5, you lose."
380 PRINT"You are in the middle of the desert";
390 PRINT". Good luck."
400 C=0:C1=0:F=0:P=0:Q=0:S=6:Z=4
410 IF LEFT$(D$,1)="N" THEN 450
420 PRINT"Hit any key to continue.";
430 A=USR(0):IF A<0 THEN 430
440 GOSUB 2130
450 IF C>99 THEN 1540
460 Z=Z-1
470 IF Z<>1 THEN 500
480 PRINT TAB(4)"===== WARNING =====";
490 PRINT"    Get a drink"
500 IF Z>0 THEN 530
510 PRINT"Sorry chum, but you forgot to ";
520 PRINT"drink the water!!":GOTO 1820
530 P=P+1:X2=INT(7*RND(1)+2.5)
540 IF Q>0 THEN 1040
550 IF P<4 THEN 660
560 C1=C1+X2:IF C1<C THEN 620
```

```
570 PRINT"The pigmies have captured you ! ";
580 PRINT"Sorry to say,"
590 PRINT"camel and people soup is their ";
600 PRINT"favourite dish."
610 GOTO 2040
620 PRINT
630 PRINT"The pigmies are"C-C1;"mile";
640 IF C-C1<>1 THEN PRINT "s";
650 PRINT" behind you."
660 PRINT
670 PRINT"You have travelled";
680 IF C<>0 THEN 700
690 PRINT" nowhere so far.":GOTO 320
700 PRINT C "mile";
710 IF C<>1 THEN PRINT "s";
720 PRINT" altogether."
730 PRINT"Your camel has";7-F;"day";
740 IF 7-F<>1 THEN PRINT "s";
750 PRINT" left."
760 PRINT"You have"S;"drink";
770 IF S<>1 THEN PRINT "s";
780 PRINT" left in your canteen."
790 PRINT"You can go"Z;"command";
800 IF Z<>1 THEN PRINT "s";
810 PRINT" without drinking."
820 PRINT
830 INPUT"What is your command ";Y$
840 Y=VAL(Y$):IF Y>=1 AND Y<=5 THEN 880
850 IF Y$="H" THEN 870
860 PRINT"Invalid command. ";:GOTO 830
870 GOSUB 2130:GOSUB 2180:GOTO 820
880 GOSUB 2130
890 ON Y GOTO 1080,960,1010,1060
900 T=INT(10*RND(1))
910 IF T<>1 THEN 1530
920 PRINT"Help has found you unconcious. ";
930 PRINT"Fortunately"
940 PRINT"they have managed to revive you."
950 S=3:Z=4:GOTO 490
960 F=F+1:IF F=8 THEN 1510
970 GOSUB 1120
980 X1=INT(10*RND(1)+1):C=C+X1
990 PRINT"Your camel likes this pace."
1000 GOTO 450
1010 F=F+3:IF F>7 THEN 1510
1020 GOSUB 1120
1030 X1=2*INT(10*RND(1)+1):C=C+X1
1040 PRINT"Your camel is burning across ";
1050 PRINT"the desert sands.":PRINT:GOTO 450
1060 PRINT"Your camel thanks you!"
1070 F=0:GOTO 460
1080 S=S-1:IF S<0 THEN 1530
1090 PRINT"You had better watch out for ";
1100 PRINT"an oasis!"
1110 Z=4:GOTO 820
1120 A=INT(100*RND(1)):IF A>5 THEN 1430
```

```
1130 PRINT"Wild berbers hidden in the sand ";
1140 PRINT"have captured"
1150 PRINT"you. Luckily the local Sheik ";
1160 PRINT"has agreed to"
1170 PRINT"their ransom demands. .... ";
1180 PRINT"But watch out"
1190 PRINT"the pigmies are catching you up."
1200 PRINT:PRINT"You have a choice:":PRINT
1210 PRINT"1 Wait while the Sheik raises ";
1220 PRINT"the money."
1230 PRINT"2 Try to escape.":PRINT
1240 INPUT"What is your choice ";X
1250 IF X=1 OR X=2 THEN 1270
1260 PRINT"Invalid. ";GOTO 1240
1270 IF X=1 THEN 1350
1280 X1=INT(10*RND(1)):IF X1<5 THEN 1320
1290 PRINT"Well done, you have escaped ";
1300 PRINT"successfully."
1310 Q=0:GOTO 450
1320 PRINT"You were mortally wounded by ";
1330 PRINT"a pig stabber"
1340 PRINT"whilst escaping.":GOTO 1820
1350 X1=INT(100*RND(1)):IF X1>24 THEN 1400
1360 PRINT"Your ransom has been paid and ";
1370 PRINT"you are free"
1380 PRINT"to go."
1390 Q=0:GOTO 450
1400 PRINT"The Sheik is collecting .... ";
1410 PRINT"just wait."
1420 GOTO 1200
1430 A=INT(10*RND(1)):IF A>2 THEN 1590
1440 PRINT"You have arrived at an oasis.";
1450 PRINT"    You have filled"
1460 PRINT"your canteen, and your camel ";
1470 PRINT"is drinking whilst"
1480 PRINT"you are eating figs."
1490 IF F>3 THEN F=2
1500 Z=4:S=6:RETURN
1510 PRINT"You dirty thief, you've ridden ";
1520 PRINT"your camel":PRINT"to death."
1530 GOTO 1820
1540 PRINT"You win, a party is being given ";
1550 PRINT"in your honour."
1560 PRINT"The pigmies have given up and ";
1570 PRINT"have gone home."
1580 GOTO 2040
1590 X1=INT(100*RND(1)):IF X1>5 THEN 1760
1600 PRINT"You have been caught in a ";
1610 PRINT"sandstorm. Good luck."
1620 X5=INT(10*RND(1)+1):X6=INT(10*RND(1))
1630 IF X6<5 THEN 1650
1640 C=C+X5:GOTO 1660
1650 C=C-X5
1660 PRINT"Your new position is"X5;"mile";
1670 IF X5<>1 THEN PRINT "s";
1680 PRINT" off your intended"
```

```
1690 PRINT"course."
1700 IF X6<5 THEN GOTO 1730
1710 PRINT"Fortunately the pigmies are now ";
1720 PRINT"further behind.":GOTO 1750
1730 PRINT"You got lost, and the pigmies ";
1740 PRINT"are now closer."
1750 RETURN
1760 X1=INT(100*RND(1)):IF X1>5 THEN RETURN
1770 C1=C1+1
1780 PRINT"Your camel hurt his hump!"
1790 PRINT"Fortunately, the pigmies are ";
1800 PRINT"foot weary."
1810 RETURN
1820 U=INT(10*RND(1))
1830 PRINT"You died in the desert."
1840 IF U>1 THEN 1890
1850 PRINT"The 'National Camel Drivers' ";
1860 PRINT"union' are not"
1870 PRINT"attending your funeral!!!"
1880 GOTO 2040
1890 IF U>3 THEN 1930
1900 PRINT"Your body has been eaten by ";
1910 PRINT"vultures!!!"
1920 GOTO 2040
1930 IF U>5 THEN 1970
1940 PRINT"The local Sheik is now using ";
1950 PRINT"your skull":PRINT"as a handbag!!!"
1960 GOTO 2040
1970 IF U>7 THEN 2020
1980 PRINT"People with little experience ";
1990 PRINT"should keep out"
2000 PRINT"of the desert!!!"
2010 GOTO 2040
2020 PRINT"Well it was pretty dumb to steal ";
2030 PRINT"that Idol":PRINT"anyway!!!"
2040 PRINT:PRINT:PRINT"Would you like ";
2050 PRINT"a new camel, and a new ";
2060 INPUT "game ";D$
2070 IF LEFT$(D$,1)<>"Y" THEN 2090
2080 D$="N":CLS:GOTO 380
2090 PRINT:PRINT:PRINT TAB(18);"CHICKEN !!!!"
2100 PRINT:END
2110 REM
2120 REM Clear screen and put up title
2130 CLS:T$="++ CAMEL ++":FOR I=1 TO LEN(T$)
2140 POKE 3034+I,ASC(MID$(T$,I,1)):NEXT
2150 RETURN
2160 REM
2170 REM List of commands
2180 PRINT"Commands:"
2190 PRINT"1 Drink from canteen";TAB(26);
2200 PRINT"4 Stop for the night
2210 PRINT"2 Ahead half speed";TAB(26);
2220 PRINT"5 Hope for help"
2230 PRINT"3 Ahead full speed";TAB(26);
2240 PRINT"H Repeat commands"
2250 RETURN
```

Ok

```
10 REM          ***** COMRADE X *****
20 REM
30 REM  Adapted for 16 K Nascom 1/2 fitted with
40 REM  NAS-SYS 1 or 'T' series monitors.
50 REM
60 REM  by D. R. Hunt. June 1979.
70 REM  From 'KING' by James A. Storer, in
80 REM  '101 Basic Computer Games' published
90 REM  by Creative Computing.
100 REM
110 CLEAR 150: CLS
120 REM
130 REM  Check monitor in use and set pointer.
140 RESTORE 200: IF PEEK(1)=0 THEN RESTORE 170
150 REM
160 REM  DATA for NAS-SYS use.
170 DATA 25055,1080,-53,536,-20665,3370,-5664,0
180 REM
190 REM  DATA for NASBUG use.
200 DATA 27085,14336,-13564,6399,18178,10927,-8179,233
210 REM  Load machine code 'INKEY' command
220 DOKE 4100,3328:FOR I9=3328 TO 3342 STEP 2
230 READ I8:DOKE I9,I8:NEXT
240 REM
250 REM  Get name
260 SCREEN 1,7:Y1$="":PRINT"W";
270 INPUT"hat is your name ";Y1$
280 IF Y1$<>"" THEN 300
290 SCREEN 1,7:PRINT"Please .... w";:GOTO 270
300 Z1=LEN(Y1$):Z2=((30-Z1)/2)
310 REM
320 REM  Convert name to lower case.
330 FOR I=1 TO LEN(Y1$):A=ASC(MID$(Y1$,I,1))
340 IF I9=1 OR I=1 THEN IFA>=96 THEN A=A-32
350 IF I9=1 OR I=1 THEN 380
360 IF A<=57 THEN 380
370 IF A<=95 THEN A=A+32
380 Y$=Y$+CHR$(A)
390 I9=0:IF A=32 THEN Y2$="":GOTO 410
400 Y2$=Y2$+CHR$(A)
410 IF A<=57 THEN I9=1
420 IF Y2$="Mc" THEN I9=1
430 IF Y2$="Mc" OR Y2$="Mac" THEN I8=1
440 NEXT
450 GOSUB 4480
460 REM
470 REM  Ask for instructions.
480 IF I8<>1 THEN 500
490 SCREEN 1,6:PRINT"A scots dictator eh !!"
500 SCREEN 1,7:Z$="":PRINT"D";
510 INPUT "o you want instructions ";Z$
520 IF Z$<>"" THEN 540
530 SCREEN 1,7:PRINT"Please .. d";:GOTO 510
540 N5=8:IF LEFT$(Z$,1)="N" THEN 880
550 IF LEFT$(Z$,1)="A" THEN 3900
560 IF LEFT$(Z$,1)<>"Y" THEN 530
```

```
570 GOSUB 4480
580 PRINT"Congratulations!   You've just ";
590 PRINT"been elected"
600 PRINT"Premier of the small communist ";
610 PRINT"island, Niatirb,"
620 PRINT"which is 70 miles by 30. Your job ";
630 PRINT"is to decide"
640 PRINT"the country's budget and ";
650 PRINT"distribute money to"
660 PRINT"your countrymen. The money ";
670 PRINT"system is rallods,"
680 PRINT"and each person needs 100 rallods ";
690 PRINT"per year to"
700 PRINT"survive. Income comes from farming ";
710 PRINT"and tourism,"
720 PRINT"half your land is farmed, and ";
730 PRINT"also has a high"
740 PRINT"mineral content which may be sold ";
750 PRINT"to foreign"
760 PRINT"industry who will import and ";
770 PRINT"support their own"
780 PRINT"workers. Crops cost between 10 and ";
790 PRINT"15 rallods"
800 PRINT"per sq. mile to plant."
810 PRINT"Your goal is to complete your ";N5;
820 PRINT"year term of"
830 PRINT"office.";TAB(35);"Good luck !!"
840 PRINT"Hit any key to continue.";
850 GOSUB 4520
860 REM
870 REM Set up first conditions
880 A=INT(60000+(1000*RND(1))-(1000*RND(1)))
890 B=INT(500+(10*RND(1))-(10*RND(1)))
900 D=2000:C=0
910 N5=8:X5=0:B1=B*.666666:GOSUB 4480
920 W=INT(10*RND(1)+95):A3=A
930 REM
940 REM Set up screen
950 PRINT"Comrade ";Y$;" you now have :";
960 PRINT N5-X5;"years left to serve"
970 PRINT A; "rallods in the treasury"
980 PRINT INT(B);"countrymen"
990 V9=INT(((RND(1)/2)*10+10))
1000 IF C=0 THEN 1020
1010 PRINT INT(C)"foreign workers"
1020 PRINT INT(D);"square miles of land."
1030 PRINT
1040 PRINT"This year industry will buy land at";
1050 PRINT W:PRINT"rallods per square mile."
1060 PRINT"Land currently costs";V9;"rallods ";
1070 PRINT"per square":PRINT"mile to plant."
1080 REM
1090 REM Get inputs for this year.
1100 N4=N4-1
1110 I$=""
1120 PRINT:PRINT"How many square miles will ";
```

```
1130 INPUT"you sell ";I$:IF I$="" THEN 1110
1140 FOR H=LEN(I$) TO 1 STEP -1
1150 IF MID$(I$,H,1)="0" THEN 1110
1160 NEXT:H=VAL(I$)
1170 IF H<0 THEN 1110
1180 IF X=0 AND H>0 AND H<=D-1000 THEN 1270
1190 IF H<=D-1000 THEN 1350
1200 PRINT:PRINT"Comrade ";Y$;" think again, ";
1210 PRINT"you only":PRINT"have"D-1000;
1220 PRINT"square mile";
1230 IF D-1000<>1 THEN PRINT "s";
1240 PRINT" of farm land."
1250 IF X<>0 AND H<D-1000 THEN 1350
1260 GOTO 1110
1270 PRINT"(Foreign industry will only buy ";
1280 PRINT"farm land as"
1290 PRINT"forest land is uneconomical to ";
1300 PRINT"strip mine due"
1310 PRINT"to trees, thick top soil, etc. ";
1320 PRINT"If you still":PRINT"wish to sell, ";
1330 PRINT"re-enter the ammount.)"
1340 X=1:GOTO 1110
1350 D=INT(D-H):A=INT(A+H*W)
1360 I$=""
1370 PRINT:PRINT"How many rallods will you ";
1380 PRINT"distribute to your"
1390 INPUT"countrymen ";I$:IF I$="" THEN 1370
1400 FOR I=LEN(I$) TO 1 STEP -1
1410 IF MID$(I$,I,1)="0" THEN 1370
1420 NEXT:I=VAL(I$)
1430 IF I<0 THEN 1370
1440 IF I<A THEN 1500
1450 IF I=A THEN 1490
1460 PRINT:PRINT"Comrade ";Y$;" think again, ";
1470 PRINT"you only":PRINT"have";A;"rallods ";
1480 PRINT"in the treasury.":GOTO 1370
1490 J=0:K=0:A=0:GOTO 2100
1500 A=INT(A-I)
1510 I$=""
1520 PRINT:PRINT"Comrade ";Y$;" how many ";
1530 PRINT"square miles"
1540 INPUT"will you plant ";I$:IFI$=""THEN 1520
1550 FOR J=LEN(I$) TO 1 STEP -1
1560 IF MID$(I$,J,1)="0" THEN 1520
1570 NEXT:J=VAL(I$)
1580 IF J<0 THEN 1520
1590 IF J<=B*2 THEN 1630
1600 PRINT:PRINT"Comrade ";Y$;" , we are sorry ";
1610 PRINT"but each":PRINT"countryman can only";
1620 PRINT" plant 2 sq. miles.":GOTO 1520
1630 IF J<=D-1000 THEN 1690
1640 PRINT:PRINT"Sorry Comrade ";Y$;" but you";
1650 PRINT" only":PRINT"have";D-1000;"square ";
1660 PRINT"mile";
1670 IF D-1000<>1 THEN PRINT "s";
1680 PRINT" of farm land.":GOTO 1520
```



```
1690 U1=INT(J*V9):IF U1<A THEN 1770
1700 IF U1=A THEN 1760
1710 PRINT:PRINT"Comrade ";Y$;" , how will you";
1720 PRINT" pay for":PRINT"the grain, you ";
1730 PRINT"only have";A;"rallod";
1740 IF A<>1 THEN PRINT "s";
1750 PRINT:PRINT"in the treasury.":GOTO 1520
1760 K=0:A=0:GOTO 2100
1770 A=A-U1
1780 I$=""
1790 PRINT:PRINT"How many rallods will you ";
1800 PRINT"spend on pollution"
1810 INPUT"control ";I$:IF I$="" THEN 1790
1820 FOR K=LEN(I$) TO 1 STEP -1
1830 IF MID$(I$,K,1)="0" THEN 1790
1840 NEXT:K=VAL(I$)
1850 IF K<0 THEN 1790
1860 IF K<=A THEN 2100
1870 PRINT:PRINT"Comrade ";Y$;" think again,"
1880 PRINT"you have only";A;"rallod";
1890 IF A<>1 THEN PRINT "s";
1900 PRINT" left."
1910 GOTO 1790
1920 REM
1930 REM Exit from program
1940 IF H<>0 THEN 2110
1950 IF I<>0 THEN 2110
1960 IF J<>0 THEN 2110
1970 IF K<>0 THEN 2110
1980 PRINT
1990 PRINT"Goodbye Comrade ";Y$
2000 PRINT"(If you wish to continue office at";
2010 PRINT" a later date":PRINT"then answer ";
2020 PRINT"'AGAIN' when asked if you want"
2030 PRINT"instructions at the start.)"
2040 END
2050 REM
2060 REM Calculate results of year inputs
2070 REM and display.
2080 REM
2090 REM Starvation deaths
2100 GOTO 1940
2110 GOSUB 4480
2120 A=INT(A-K):A4=A
2130 IF A4/A3>.2 THEN 3590
2140 IF INT(I/100-B)>=0 THEN 2200
2150 IF I/100<50 THEN 3510
2160 PRINT INT(B-(I/100));"countrymen died";
2170 PRINT" of starvation."
2180 REM
2190 REM Polution deaths
2200 F1=INT(RND(1)*(2000-D))
2210 IF K<25 THEN 2230
2220 F1=INT(F1/(K/25))
2230 IF F1<=0 THEN 2280
2240 PRINT F1;"countrymen died of pollution ";
```

```
2250 PRINT"poisoning."
2260 REM
2270 REM Funeral expenses
2280 IF INT((I/100)-B)<0 THEN 2310
2290 IF F1>0 THEN 2370
2300 GOTO 2490
2310 PRINT"You were forced to spend";
2320 PRINT INT((F1+(B-(I/100)))*9);"rallods";
2330 PRINT" on":PRINT"funeral expenses."
2340 B5=INT(F1+(B-(I/100)))
2350 A=INT(A-((F1+(B-(I/100)))*9))
2360 GOTO 2410
2370 PRINT"You were forced to spend";
2380 PRINT INT(F1*9);"rallods on"
2390 PRINT"funeral expenses.":B5=F1
2400 A=INT(A-(F1*9))
2410 IF A>=0 THEN 2480
2420 PRINT"You had insufficient reserves to";
2430 PRINT" cover the":PRINT"cost - land ";
2440 PRINT"was sold.":D=INT(D+(A/W))
2450 REM
2460 REM Arrival of workers and countrymen
2470 A=0
2480 B=INT(B-B5)
2490 IF H=0 THEN 2550
2500 C1=INT(H+(RND(1)*10)-(RND(1)*20))
2510 IF C>0 THEN 2530
2520 C1=C1+20
2530 PRINT"Some";C1;"workers came to the ";
2540 PRINT"island and"
2550 P1=INT(((I/100-B)/10)+(K/25)-((2000-D)/50)-(F1/2))
2560 PRINT ABS(P1);"countrymen ";
2570 IF P1<0 THEN 2590
2580 PRINT"came to";GOTO 2600
2590 PRINT"left";
2600 PRINT" the island."
2610 REM
2620 REM Harvest
2630 B=INT(B+P1):C=INT(C+C1)
2640 U2=INT((2000-D)*RND(1)+1.5/2)
2650 IF C=0 THEN 2670
2660 PRINT"Of";INT(J);"sq. miles planted."
2670 IF J>U2 THEN 2690
2680 U2=J
2690 PRINT"You have harvested";
2700 PRINT INT(J-U2);"sq. miles of crops."
2710 IF U2=0 THEN 2780
2720 IF T1>=2 THEN 2780
2730 PRINT"(Decrease due to ";
2740 IF T1=0 THEN 2760
2750 PRINT"increased ";
2760 PRINT"air and water pollution from "
2770 PRINT"foreign industry.) ";
2780 IF J=0 THEN Q=0:GOTO 2810
2790 Q=INT((J-2)*(W/2))
2800 PRINT"Making";INT(Q);"rallods."
```

```
2810 A=INT(A+Q)
2820 REM
2830 REM Profits from tourists
2840 V1=INT(((B-P)*22)+(RND(1)*500))
2850 V2=INT((2000-D)*15)
2860 PRINT"You made";ABS(INT(V1-V2));
2870 PRINT"rallods from the tourists."
2880 IF V2=0 THEN 2970
2890 IF V1-V2>=V3 THEN 2970
2900 PRINT"Decrease because ";
2910 G1=10*RND(1)
2920 IF G1<=2 THEN 2990
2930 IF G1<=4 THEN 3020
2940 IF G1<=6 THEN 3050
2950 IF G1<=8 THEN 3080
2960 IF G1<=10 THEN 3110
2970 V3=ABS(INT(V1-V2)):A=A+V3
2980 GOTO 3160
2990 PRINT"the fish population has"
3000 PRINT"dwindled due to water pollution.";
3010 GOTO 2970
3020 PRINT"air pollution is killing"
3030 PRINT"the game bird population.";
3040 GOTO 2970
3050 PRINT"mineral baths are being"
3060 PRINT"ruined by water pollution.";
3070 GOTO 2970
3080 PRINT"unpleasant smog is"
3090 PRINT"discouraging sunbathers.";
3100 GOTO 2970
3110 PRINT"hotels are looking shabby"
3120 PRINT"due to smog grit.";
3130 GOTO 2970
3140 REM
3150 REM Mismanagement results
3160 GOSUB 4520:GOSUB 4480:IF B5>200 THEN 3340
3170 IF B<B1 THEN 3510
3180 IF A4>500 THEN 3560
3190 IF C>B THEN 3220
3200 IF N5-1=X5 THEN 3730
3210 GOTO 4160
3220 PRINT:PRINT"The number of foreign ";
3230 PRINT"workers has exceeded":PRINT"the ";
3240 PRINT"number of countrymen, as a ";
3250 PRINT"majority they":PRINT"have ";
3260 PRINT"revolted and taken over the country.
3270 IF RND(1)<=.5 THEN 3320
3280 PRINT"You have been thrown out of ";
3290 PRINT"office and you":PRINT"are now ";
3300 PRINT"residing in prison.";
3310 GOTO 4210
3320 PRINT:PRINT"You have been assassinated !"
3330 PRINT:PRINT:PRINT"End of game.":END
3340 PRINT B5;"countrymen have died in ";
3350 PRINT"one year !!":PRINT"Due to this ";
3360 PRINT"extreme mismanagement you have"
```

```
3370 PRINT"not only been impeached and ";
3380 PRINT"thrown out of":PRINT"office but ";
3390 PRINT"you ";:M6=INT(RND(1)*10)
3400 IF M6<=3 THEN 3430
3410 IF M6<=6 THEN 3450
3420 IF M6<=10 THEN 3480
3430 PRINT"also had your left eye gouged"
3440 PRINT"out.":GOTO 4210
3450 PRINT"have gained a rotten reputation"
3460 PRINT"and are now unemployable.";
3470 GOTO 4210
3480 PRINT"have been given ,he 'Order Of"
3490 PRINT"The Boot' and are now ";
3500 GOTO 3300
3510 PRINT"Over one third of the population";
3520 PRINT" has died":PRINT"since you were ";
3530 PRINT"elected to office. The people"
3540 PRINT"(remaining) hate your guts !!"
3550 GOTO 3270
3560 IF B5-F1<2 THEN 3190
3570 PRINT"A number of your countrymen died ";
3580 PRINT"as some"
3590 PRINT A4;"rallods were left over in the ";
3600 PRINT"treasury":PRINT"and not spent to ";
3610 PRINT"the good of the people. As a"
3620 PRINT"result of this blatant covetness ";
3630 PRINT"you have been":PRINT"forced to ";
3640 PRINT"resign or commit suicide. The ";
3650 PRINT"choice":PRINT"is yours. ";
3660 PRINT"If you choose the latter, please "
3670 PRINT"turn off the computer before ";
3680 PRINT"proceeding. The":PRINT"blood ";
3690 PRINT"could cause a short circuit !!";
3700 GOTO 4210
3710 REM
3720 REM Win message
3730 PRINT"Congratulations !!!!!!!!!!"
3740 PRINT"You have sucessfully completed ";
3750 PRINT"your";N5;"year":PRINT"term ";
3760 PRINT"of office. You were extremely lucky."
3770 PRINT"None the less, it's quite an ";
3780 PRINT"achievement.":PRINT"Goodbye and ";
3790 PRINT"good luck - but then you must ";
3800 PRINT"be":PRINT"the lucky type - to ";
3810 PRINT"win this game."
3820 PRINT:Z3=1
3830 PRINT"You are eligible for re-election. ";
3840 PRINT"Do you wish ":PRINT"to see how ";
3850 PRINT"your countrymen love you ";
3860 GOTO 4250
3870 GOTO 4210
3880 REM
3890 REM Restart game here
3900 GOSUB 4480
3910 PRINT:PRINT"How many years had ";
3920 PRINT"you held office ";;INPUT X5
```

```
3930 IF X5<0 THEN 4210
3940 IF X5<N5 THEN 3980
3950 PRINT"Come on, the maximum term is only";
3960 PRINT N5;"years."
3970 GOTO 3910
3980 PRINT:PRINT"How much was in the ";
3990 PRINT"treasury ";;INPUT A
4000 IF A<0 THEN 3330
4010 PRINT:PRINT"How many countrymen ";
4020 INPUT B
4030 IF B<0 THEN 3330
4040 PRINT:PRINT"How many workers ";
4050 INPUT C
4060 IF C<0 THEN 3330
4070 PRINT:PRINT"How many square miles of ";
4080 PRINT"land ";;INPUT D
4090 IF D<0 THEN 3330
4100 IF D>2000 THEN 4120
4110 IF D>1000 THEN GOSUB 4480:GOTO 920
4120 PRINT"Don't cheat, you started with 1000";
4130 PRINT" sq. miles":PRINT"of farm land";
4140 PRINT", and 1000 sq. miles of forest."
4150 GOTO 4070
4160 X5=X5+1
4170 B5=0
4180 GOTO 920
4190 REM
4200 REM Election
4210 PRINT:PRINT"Fortunately you are eligible ";
4220 PRINT"for re-election.":PRINT"Do you ";
4230 PRINT"wish to see if your countrymen ";
4240 PRINT"have":PRINT"forgiven you ";
4250 INPUT Z$
4260 IF LEFT$(Z$,1)="Y" THEN 4280
4270 GOTO 3330
4280 GOSUB 4480:PRINT"Here are the ";
4290 PRINT"election results:"
4300 A9=(INT(RND(1)*500)+200)/10
4310 B9=(INT(RND(1)*500))/10
4320 IF Z3=1 THEN B9=B9+40:GOTO 4340
4330 B9=B9+10
4340 PRINT A9;"% of your countrymen came to ";
4350 PRINT"vote.":PRINT B9;"% of those voted ";
4360 PRINT"for you, so you are"
4370 IF B9>50 THEN 4430
4380 IF Z3=1 THEN 4440
4390 PRINT"still in political exile and ";
4400 PRINT"your ex-countrymen":PRINT"hate ";
4410 PRINT"your guts (even when fried) !!"
4420 GOTO 3330
4430 PRINT"duly re-elected.":GOSUB 4520:GOTO 910
4440 PRINT"not as well loved as you think ";
4450 PRINT"you should be !":GOTO 3330
4460 REM
4470 REM Centre title
4480 PRINT:CLS:Z1$="** COMRADE ":Z2$="**"
```

```
4490 Z$=Z1$+Y1$+Z2$:FOR Z=1TO14+Z1
4500 POKE Z+3018+Z2,ASC(MID$(Z$,Z,1)):NEXT
4510 RETURN
4520 Z=USR(0):IF Z<0 THEN 4520
4530 RETURN
4540 END
Ok
```

```
10 REM      ****      INPUT      ****
20 REM
30 REM  Program to return an input string
40 REM  equivalent to INPUT I$ but able to
50 REM  accept any input character without
60 REM  '?Extra ignored' errors.
70 REM
80 REM **
90 REM Put down machine code for keyboard input
100 FOR A=3328 TO 3334 STEP 2
110 READ B:DOKE A,B:NEXT
120 REM **
130 REM Call the routine to get the input
140 I$="":GOSUB 210
150 REM **
160 REM Now print the input, and go round again
170 PRINT I$
180 GOTO 140
190 REM **
200 REM The input routine
210 DOKE 4100,3328:I=USR(0)
220 IF I=13 THEN PRINT:RETURN
230 IF I<=7 OR I>=9 AND I<30 THEN 210
240 IF I$="" AND I=8 THEN 210
250 PRINT CHR$(I);
260 IF I=8 THEN 280
270 I$=I$+CHR$(I):GOTO 210
280 I$=LEFT$(I$,LEN(I$)-1):GOTO 210
290 REM **
300 REM **
310 REM **
320 REM **
330 REM **
340 REM ** Machine code data
350 DATA 31711,-20665, 3370,-5664
Ok
```

```

10 REM      ****   PRINT USING      ****
20 REM
30 REM A   generalized   routine which will return
40 REM an input variable as a string in the form
50 REM determined by PU$.  Thus if PU$="XXXX.XX"
60 REM then:
70 REM
80 REM   Input variable (P1)   Output string (P1$)
90 REM       1.23                      1.23
100 REM      10.5                      10.50
110 REM      9999                      9999.00
120 REM **
130 REM Get an input
140 INPUT "What input ";P1
150 REM **
160 REM Validate input
170 IF P1<-9999.99 OR P1>9999.99 THEN 140
180 REM **
190 REM Send it to the 'PRINT USING' routine
200 PU$="XXXX.XX":GOSUB 290
210 REM **
220 REM Print the result with a '#' sign
230 PRINT "#";P1$,P1
240 REM **
250 REM Go and get another input
260 GOTO 140
270 REM **
280 REM 'PRINT USING' routine
290 P1$="":FOR P2=1 TO LEN(PU$)
300 IF MID$(PU$,P2,1)<>"." THEN NEXT
310 P2$=STR$(P1):FOR P3=1 TO LEN(P2$)
320 IF MID$(P2$,P3,1)<>"." THEN NEXT
330 P2=P2-1:P3=P3-1:P4=P2-P3
340 IF P4>-1 THEN P1$=P1$+" ":P4=P4-1:GOTO 340
350 P1$=P1$+LEFT$(P2$,P3)+"."
360 FOR P4=2 TO LEN(PU$)-P2
370 P3$=MID$(P2$,P3+P4,1):IF P3$="" THEN P3$="0"
380 P1$=P1$+P3$:NEXT
390 P3="":P2$="":FOR P2=1 TO LEN(P1$)
400 P3$=MID$(P1$,P2,1)
410 IF P3$="-" THEN P3=1:P3$=" "
420 P2$=P2$+P3$:NEXT
430 P1=VAL(P2$):IF P3=1 THEN P1=P1-P1-P1
440 P2$="":P3$="":RETURN
Ok

```

Appendix =====

Many of the programs in this book use the same routine for loading the title to the top line of the screen, namely:

```
10 T$="THE TITLE"  
20 FOR T=1 TO LEN(T$)  
30 POKE 3017+T,ASC(MID$(T$,T,1)):NEXT
```

It has been noticed that some Nascom 2s running at 4MHz fail during this routine, returning with an FC Error in 30. Strangely, this is not a software bug, but a minor hardware mistiming fault. This fault may usually be cured by wiring a 220R resistor between pins 6 and 14 of IC11. To test for this fault, use the above program, adding line 40:

```
40 CLS:GOTO 20
```

and run this for some time allowing the Nascom to become thoroughly warm. If the program fails, add the resistor.

There is a minor bug in the BASIC concerning TAB and SCREEN statements, and under certain circumstances when manipulating graphics using the SET, RESET and POINT statements in conjunction with the SCREEN statement. Remember the Microsoft Basic was originally designed to be used with a printer. As a safety feature an automatic linefeed was incorporated to ensure that the print head did not overrun the print width. Internally there is a pointer which keeps track of the position of the print head (see POS statement), starting at 1 for the first character position etc. Now when the SCREEN statement is used, or the TAB statement is used incorrectly, the position of the cursor on the screen is changed without changing the value of the pointer. Hence, when more 'printing' is required, this pointer continues to increment even though the cursor may well be set back to the start of the line. When the pointer exceeds the preset width the BASIC puts out an automatic linefeed, which puts the cursor on the next line. An easy way to 'fix' this is to use a PRINT statement before any SCREEN statement thus:

```
PRINT:SCREEN X,Y
```

this always returns the width pointer to 1 before the cursor is moved. Remember the POS statement is meaningless when SCREEN statements are being used.

In several instances the length of the lines in the programs exceeds the maximum 48 characters that NAS-SYS is able to accept as an input line. (NAS-BUG will accept 76 characters.) However, NAS-SYS will behave as NAS-BUG when in the 'teletype mode', and is capable of accepting input lines up to the maximum of 76. The penalty is the loss of NAS-SYS editing. To enter a long line, enter MONITOR, then type X0, followed by a newline, then re-enter BASIC using the Z command. The BASIC is now in 'teletype mode'. To leave 'teletype mode', enter MONITOR, then type N, newline, and re-enter BASIC using the Z command. Do not forget the shorthand for PRINT is a '?'.

Mention is made in the Basic Manual of 'Listing to Tape'. Using the existing system monitors, NAS-BUG and NAS-SYS 1, this is not so easy as it sounds, as, if the 'teletype mode', X, is used, the data is sent to the tape in one continuous stream. On replay there is insufficient time after a carriage return for the BASIC to load the incoming data into the Basic Source File before the next character

arrives. The NULL command in the BASIC is intended to be used under these circumstances, but as the system monitors throw away NULLs, the null delays at the end of each line are too short to be of use. One way round this problem is to use the thoughtfully provided U command in NAS-SYS, or to change \$CRT in NAS-BUG. In each case the output routines should be directed to a small machine code routine which will call SRLOUT before going to the CRT routine. In this way the NULLs are not lost, and 'Listing to Tape' is possible.

There is only one graphics program in this book, others will follow in later editions. When graphics are listed, at first sight something very odd occurs. For instance, if a PRINT statement contains a dotted white line (using 6 "■" characters, CHR\$(195), character C3H):

```
PRINT"■■■■■■"
```

when LISTed this will appear as:

```
PRINT"TANTANTANTANTAN"
```

This is because the HEX character C3 is the code used by BASIC to denote a TAN function. This is known as a 'Reserved Word'. When the BASIC LISTs a program, it has no way of knowing (without running the program for SN Errors) if the programmer meant TAN or a graphic character. It therefore errs on the side of safety, and LISTs TAN. You and I know that within a PRINT statement, the use of a reserved word is most likely to be a graphic. So if you see a reserved word within a PRINT statement, and the context seems to imply a graphic, then a graphic character must be entered. Note that it is no use entering reserved words in a PRINT statement, as you will get back exactly what you entered. Thus if you entered:

```
PRINT "TANTANTANTANTAN"
```

it will be printed as

```
TANTANTANTANTAN
```

If you mean a graphics character, you must enter a graphics character.

Sorting out the reserved words into graphics is not an easy job. A reference table is given in the BASIC manual, but this is incomplete. Additionally, there are only 80 reserved words, the remainder are printed as alphabetic characters with some unprintable characters thrown in for good measure. A complete table is given at the end of this book. Finally, one last thing about the graphics: the last 64 characters (CHR\$(192), C0H to CHR\$(255), FFH) are the block graphics used by the SET, RESET, and POINT statements. Note that one character space is divided into 6; if the top left is given the numerical value 1, the middle left the value 2, the bottom left the value 4, the top right the value 8 etc, then it will be seen that the way the blocks fill a character space is in the form of a binary progression. It can be very useful to remember that these characters can be calculated if required.

Reserved Words, and Graphics

=====

The HEX codes for the reserved words and graphics are tabulated below. Unfortunately the results LISTed for the last 48 graphics, where there are no reserved words, are in many cases beyond the scope of the printer. Where an unprintable character is encountered, a '.' is substituted. It is suggested that users complete this table for themselves, typing in a Basic line using the combination of keys given, then LISTing the results. The key to the keys used is:

Gr = Graph, Ct = CTRL, Sh = SHIFT

The keys should be depressed in the order listed.

HEX code	Basic equiv.	Keys	Reserved Word
80H	CHR\$(128)	Gr/Ct/Sh/3	= END
81H	CHR\$(129)	Gr/Ct/A	= FOR
82H	CHR\$(130)	Gr/Ct/B	= NEXT
83H	CHR\$(131)	Gr/Ct/C	= DATA
84H	CHR\$(132)	Gr/Ct/D	= INPUT
85H	CHR\$(133)	Gr/Ct/E	= DIM
86H	CHR\$(134)	Gr/Ct/F	= READ
87H	CHR\$(135)	Gr/Ct/G	= LET
88H	CHR\$(136)	Gr/Ct/H	= GOTO
89H	CHR\$(137)	Gr/Ct/I	= RUN
8AH	CHR\$(138)	Gr/Ct/J	= IF
8BH	CHR\$(139)	Gr/Ct/K	= RESTORE
8CH	CHR\$(140)	Gr/Ct/L	= GOSUB
8DH	CHR\$(141)	Gr/Ct/M	= RETURN
8EH	CHR\$(142)	Gr/Ct/N	= REM
8FH	CHR\$(143)	Gr/Ct/O	= STOP
90H	CHR\$(144)	Gr/Ct/P	= OUT
91H	CHR\$(145)	Gr/Ct/Q	= ON
92H	CHR\$(146)	Gr/Ct/R	= NULL
93H	CHR\$(147)	Gr/Ct/S	= WAIT
94H	CHR\$(148)	Gr/Ct/T	= DEF
95H	CHR\$(149)	Gr/Ct/U	= POKE
96H	CHR\$(150)	Gr/Ct/V	= DOKE
97H	CHR\$(151)	Gr/Ct/W	= SCREEN
98H	CHR\$(152)	Gr/Ct/X	= LINES
99H	CHR\$(153)	Gr/Ct/Y	= CLS
9AH	CHR\$(154)	Gr/Ct/Z	= WIDTH
9BH	CHR\$(155)	Gr/Ct/[= MONITOR
9CH	CHR\$(156)	Gr/Ct/Sh/[= SET
9DH	CHR\$(157)	Gr/Ct/]	= RESET
9EH	CHR\$(158)	Gr/Ct/Sh/0	= PRINT
9FH	CHR\$(159)	Gr/Ct/Sh/]	= CONT
A0H	CHR\$(160)	Gr/space	= LIST
A1H	CHR\$(161)	Gr/Sh/1	= CLEAR
A2H	CHR\$(162)	Gr/Sh/2	= CLOAD
A3H	CHR\$(163)	Gr/Sh/3	= CSAVE
A4H	CHR\$(164)	Gr/Sh/4	= NEW
A5H	CHR\$(165)	Gr/Sh/5	= TAB(
A6H	CHR\$(166)	Gr/Sh/6	= TO
A7H	CHR\$(167)	Gr/Sh/7	= FN
A8H	CHR\$(168)	Gr/Sh/8	= SPC(

A9H	CHR\$(169)	Gr/Sh/9	=	THEN
AAH	CHR\$(170)	Gr/Sh/:	=	NOT
ABH	CHR\$(171)	Gr/Sh/;	=	STOP
ACH	CHR\$(172)	Gr/Sh/,	=	+
ADH	CHR\$(173)	Gr/-	=	-
AEH	CHR\$(174)	Gr/.	=	*
AFH	CHR\$(175)	Gr//	=	/
B0H	CHR\$(176)	Gr/0	=	.
B1H	CHR\$(177)	Gr/1	=	AND
B2H	CHR\$(178)	Gr/2	=	OR
B3H	CHR\$(179)	Gr/3	=	>
B4H	CHR\$(180)	Gr/4	=	=
B5H	CHR\$(181)	Gr/5	=	<
B6H	CHR\$(182)	Gr/6	=	SGN
B7H	CHR\$(183)	Gr/7	=	INT
B8H	CHR\$(184)	Gr/8	=	ABS
B9H	CHR\$(185)	Gr/9	=	USR
BAH	CHR\$(186)	Gr/:	=	FRE
BBH	CHR\$(187)	Gr/;	=	INP
BCH	CHR\$(188)	Gr/Sh/,	=	POS
BDH	CHR\$(189)	Gr/Sh/-	=	SQR
BEH	CHR\$(190)	Gr/Sh/.	=	RND
BFH	CHR\$(191)	Gr/Sh//	=	LOG
C0H	CHR\$(192)	Gr/Sh/@	=	EXP
C1H	CHR\$(193)	Gr/A	=	COS
C2H	CHR\$(194)	Gr/B	=	SIN
C3H	CHR\$(195)	Gr/C	=	TAN
C4H	CHR\$(196)	Gr/D	=	ATN
C5H	CHR\$(197)	Gr/E	=	PEEK
C6H	CHR\$(198)	Gr/F	=	DEEK
C7H	CHR\$(199)	Gr/G	=	POINT
C8H	CHR\$(200)	Gr/H	=	LEN
C9H	CHR\$(201)	Gr/I	=	STR\$
CAH	CHR\$(202)	Gr/J	=	VAL
CBH	CHR\$(203)	Gr/K	=	ASC
CCH	CHR\$(204)	Gr/L	=	CHR\$
CDH	CHR\$(205)	Gr/M	=	LEFT\$
CEH	CHR\$(206)	Gr/N	=	RIGHT\$
CFH	CHR\$(207)	Gr/O	=	MID\$
End of list of reserved words.				
D0H	CHR\$(208)	Gr/P	=	r
D1H	CHR\$(209)	Gr/Q	=	hy
D2H	CHR\$(210)	Gr/R	=	g
D3H	CHR\$(211)	Gr/S	=	v
D4H	CHR\$(212)	Gr/T	=	lp
D5H	CHR\$(213)	Gr/U	=	j
D6H	CHR\$(214)	Gr/V	=	.
D7H	CHR\$(215)	Gr/W	=	k(
D8H	CHR\$(216)	Gr/X	=	o,
D9H	CHR\$(217)	Gr/Y	=	1
DAH	CHR\$(218)	Gr/Z	=	.
DBH	CHR\$(219)	Gr/[=	j-
DCH	CHR\$(220)	Gr/Sh/\	=	j.
DDH	CHR\$(221)	Gr/]	=	j
DEH	CHR\$(222)	Gr/Sh/0	=	.
DFH	CHR\$(223)	Gr/Sh/]	=	jF

E0H	CHR\$(224)	Gr/Ct/Sh/space	=	h.
E1H	CHR\$(225)	Gr/Sh/A	=	jK
E2H	CHR\$(226)	Gr/Sh/B	=	jr
E3H	CHR\$(227)	Gr/Sh/C	=	jp
E4H	CHR\$(228)	Gr/Sh/D	=	hM
E5H	CHR\$(229)	Gr/Sh/E	=	t
E6H	CHR\$(230)	Gr/Sh/F	=	a
E7H	CHR\$(231)	Gr/Sh/G	=	j
E8H	CHR\$(232)	Gr/Sh/H	=	l
E9H	CHR\$(233)	Gr/Sh/I	=	hS
EAH	CHR\$(234)	Gr/Sh/J	=	t✓
EBH	CHR\$(235)	Gr/Sh/K	=	q
ECH	CHR\$(236)	Gr/Sh/L	=	*
EDH	CHR\$(237)	Gr/Sh/M	=	u
EEH	CHR\$(238)	Gr/Sh/N	=	G
EFH	CHR\$(239)	Gr/Sh/O	=	.
F0H	CHR\$(240)	Gr/Sh/P	=	f
F1H	CHR\$(241)	Gr/Sh/Q	=	.
F2H	CHR\$(242)	Gr/Sh/R	=	-
F3H	CHR\$(243)	Gr/Sh/S	=	.
F4H	CHR\$(244)	Gr/Sh/T	=	.
F5H	CHR\$(245)	Gr/Sh/U	=	.
F6H	CHR\$(246)	Gr/Sh/V	=	%
F7H	CHR\$(247)	Gr/Sh/W	=	.
F8H	CHR\$(248)	Gr/Sh/X	=	"
F9H	CHR\$(249)	Gr/Sh/Y	=	.T.W.#
FAH	CHR\$(250)	Gr/Sh/Z	=	k
FBH	CHR\$(251)	Gr/Ct/;	=	.
FCH	CHR\$(252)	Gr/Ct/Sh/,	=	h
FDH	CHR\$(253)	Gr/Ct/Sh/-	=	.
FEH	CHR\$(254)	Gr/Ct/Sh/.	=	f
FFH	CHR\$(255)	Gr/Ct/Sh//	=	J

10000 END
=====

Well that's all folks, we hope that you have enjoyed the preceeding progs. and that they were all bug free (they should be, we've tried them all as best we can).

This booklet was produced by the INMC whose other claim to literary fame is the INMC News magazines. The INMC News is produced every 6-8 weeks and contains much useful information to the NASCOM owner. If you are not already a member and are interested in finding out more then write to us at the address given on the inside front cover of this remarkable publication (We're not biased - HONEST).

Happy programming.