

4.2.9 PROGRAM 9

This program allows a screenful of text to be created and edited on the screen and then saved on cassette tape. The program begins by clearing the screen, lines 150-160, using CRT to display the 'clear screen' character, OCH. (A better alternative here is to program RST PRS:DEFB OCH,0. This requires only three bytes instead of four.) Next, at lines 180-190, the input table is set so that input is from the keyboard only. This avoids any spurious inputs from the cassette recorder when rewinding, etc.

At TEXT, line 210, the cursor is blinked until a keyboard character is input. Either alphanumeric characters or cursor control keys may be entered to produce the required display. If the input character is a NUL (CNTRL/SHT/0), the screen editing is terminated and the program jumps to TAPE. Otherwise, the input character is displayed and the process repeated.

Saving the screen text on cassette tape, TAPE at line 280, begins by transferring the video ram to memory locations A000 to A400. It is these locations which will actually be written to the tape. This is necessary because during the write to tape information indicating the progress of the Write command is displayed and this information corrupts the desired screenful of text.

At line 330, a W A000 A400 command is programmed. Workspace location ARG1 is loaded with A000, location ARG2 is loaded with A400 and then the W command is called as SCAL "W". At this time the cassette recorder should be set to record. The program ends by returning control to NAS-SYS.

The tape so saved may, at some later time, be read back using the R command and the required text again displayed by copying from A000 to 0800; the command string is thus

C A000 0800 0400

Note that if the cassette input of a second Nascom doing an R command were to be connected to the Nascom running this program, the second Nascom could receive the screenful of text. Since the cassette output is a sequence of audio tones, any audio link, such as the telephone system or radio, may be used. This program could therefore be used as a simple method of communicating between Nascoms. Of course, the legal restrictions applying to the communication medium will have to be observed.

Listing of Program 9

```

          0010 ;PROGRAM 9
          0020 ;
2D00      0030      ORG 2D00H
          0040 ;
          0050 ;NAS-SYS routine numbers.
2D00 007B  0060 BLINK EQU 7BH
2D00 0065  0070 CRT   EQU 65H
2D00 0072  0080 NIM   EQU 72H
2D00 005B  0090 MRET  EQU 5BH
          0100 ;NAS-SYS workspace
2D00 0C0C  0110 ARG1  EQU 0C0CH
2D00 0C0E  0120 ARG2  EQU 0C0EH
2D00 0780  0130 INTAB EQU 0780H ;For NAS-SYS 1,0786H.
          0140 ;
2D00 3E0C  0150      LD A,0CH    ;Clear the
2D02 DF65  0160      SCAL CRT    ; screen.
          0170 ;
2D04 218007 0180      LD HL,INTAB ;Input table is
2D07 DF72  0190      SCAL NIM   ; keyboard only.
          0200 ;Enter and edit text.
2D09 DF7B  0210 TEXT   SCAL BLINK ;Get a key
2D0B A7   0220      AND A       ;Is it CTRL/SHTF/@
2DOC 2804  0230      JR Z,TAPE   ;If so, jump, else..
2D0E DF65  0240      SCAL CRT    ; display character.
2D10 18F7  0250      JR TEXT
          0260 ;Transfer to tape.
          0270 ;First copy video ram to A000-A400.
2D12 210008 0280 TAPE   LD HL,0800H
2D15 1100A0  0290      LD DE,0A000H
2D18 010004  0300      LD BC,0400H
2D1B EDB0  0310      LDIR
          0320 ;Do a W A000 A400
2D1D 2100A0  0330      LD HL,0A000H
2D20 220C0C  0340      LD (ARG1),HL
2D23 2100A4  0350      LD HL,0A400H
2D26 220E0C  0360      LD (ARG2),HL
2D29 DF57  0370      SCAL "W"
          0380 ;
2D2B DF5B  0390      SCAL MRET

```

65970
SMW

4.10 PROGRAM 10

This program is a development of Program 9. The tape of screen data when replayed will be automatically displayed on the screen. This is achieved by writing onto the tape itself the commands to read the tape into the screen buffer and copy from the screen buffer to the video ram. (This is similar to the technique used by the Generate command.)

The first part of the program, lines 200 to 300, is the text editing process, as in Program 9. When the data is to be saved, the video ram contents are copied into the buffer area which starts at location BUF. At line 390, the output table is defined to avoid output to any devices other than the screen and tape.

Next, the sequence (CR)

E0 (CR)

R (CR)

is written to the tape by lines 430 to 540. A delay gives time for the Nascom to respond following receipt of these commands from the tape. The text data is then written using a call to the W command, lines 570 to 610. Following another short delay the command string for a COPY command is written to the tape. The original output table is restored in lines 760 to 770.

Listing of Program 10

```

          0010 ;PROGRAM 10
          0020 ;
2D00      0030      ORG 2D00H
          0040 ;NAS-SYS routine numbers.
2D00 007B  0050 BLINK EQU 7BH
2D00 0065  0060 CRT   EQU 65H
2D00 0072  0070 NIM   EQU 72H
2D00 0071  0080 NOM   EQU 71H
2D00 005B  0090 MRET  EQU 5BH
2D00 0066  0100 TBCD3 EQU 66H
2D00 0028  0110 PRS   EQU 28H
2D00 0038  0120 RDEL  EQU 38H
2D00 0030  0130 ROUT  EQU 30H
          0140 ;NAS-SYS workspace
2D00 0C0C  0150 ARG1  EQU 0C0CH
2D00 0C0E  0160 ARG2  EQU 0C0EH
2D00 0780  0170 INTAB EQU 0780H ;For NAS-SYS 1,0786H.
2D00 0774  0180 OUTT2 EQU 0774H
          0190 ;
2D00 3E0C  0200 START LD A,0CH ;Clear the
2D02 DF65  0210 SCAL CRT   ; screen.
          0220 ;
2D04 218007 0230 LD HL,INTAB ;Input table is
2D07 DF72  0240 SCAL NIM   ; keyboard only.
          0250 ;Enter and edit text.
2D09 DF7B  0260 TEXT  SCAL BLINK ;Get a key
2D0B A7    0270 AND A      ;Is it CTRL/SHT/@
2D0C 2804  0280 JR Z,TAPE ;If so, jump, else..
2D0E DF65  0290 SCAL CRT   ; display character.
2D10 18F7  0300 JR TEXT

```

A Guide to NAS-SYS

Listing of Program #10, continued

```

0310 ;Transfer to tape.
0320 ;First copy video ram BUF.
2D12 210008 0330 TAPE LD HL,0800H
2D15 11642D 0340 LD DE,BUF
2D18 010004 0350 LD BC,0400H
2D1B EDB0 0360 LDIR
0370 ;
0380 ;Set output table to CRT & SRLX.
2D1D 217407 0390 LD HL,OUTT2
2D20 DF71 0400 SCAL NOM
2D22 E5 0410 PUSH HL ;Save previous table.
0420 ;Output the command header.
2D23 215E2D 0430 LD HL,ECS
2D26 0606 0440 LD B,6
2D28 7E 0450 OC LD A,(HL)
2D29 F7 0460 RST ROUT
0470 ;Wait
2D2A 0E20 0480 LD C,20H
2D2C AF 0490 XOR A
2D2D FF 0500 W1 RST RDEL
2D2E 0D 0510 DEC C
2D2F 20FC 0520 JR NZ,W1
2D31 23 0530 INC HL
2D32 10F4 0540 DJNZ OC
0550 ;Output text data.
0560 ;Do a W buf buf+400 BUF ALREADY
2D34 21642D 0570 LD HL,BUF
2D37 220C0C 0580 LD (ARG1),HL
2D3A 216431 0590 LD HL,BUF+400H
2D3D 220E0C 0600 LD (ARG2),HL
2D40 DF57 0610 SCAL "W
0620 ;
0630 ;Wait
2D42 AF 0640 XOR A
2D43 FF 0650 RST RDEL
0660 ;
0670 ;Output C buffer address 0800 0400.
2D44 3E43 0680 LD A,"C
2D46 F7 0690 RST ROUT
2D47 21642D 0700 LD HL,BUF
2D4A DF66 0710 SCAL TBCD3
2D4C EF 0720 RST PRS
2D4D 30383030 0730 DEFN /0800 0400/
20303430
30
2D56 0D00 0740 DEFB ODH,0
0750 ;
2D58 E1 0760 POP HL
2D59 DF71 0770 SCAL NOM
0780 ;
2D5B C3002D 0790 JP START
2D5E 0D45300D 0800 ECS DEFN ODH,"E,"0,ODH,"R,ODH
520D
2D64 0400 0810 BUF DEFS 400H

```

4/16 4-20

4.11 PROGRAM 11

This program simply inputs characters from the keyboard and outputs them to the video display and the serial output port. The repeat keyboard facility is not used.

Lines 100 and 110 form a loop which is terminated only when an input character from the keyboard is detected. This character, in register A, is displayed and transmitted from the serial port by SRLX and the process repeats.

When the tape is replayed after resetting the Nascom, it will have the same effect as a phantom pressing the keyboard.

Listins of Program 11

```
0010 ;PROGRAM 11
0020 ;
2D00      0030      ORG 2D00H
0040 ;
0050 ;NAS-SYS routine numbers:
2D00 0065 0060 CRT EQU 65H
2D00 0061 0070 KBD EQU 61H
2D00 006F 0080 SRLX EQU 6FH
0090 ;
2D00 DF61 0100 LOOP SCAL KBD ;Wait for a
2D02 30FC 0110 JR NC,LOOP ; keyboard input.
2D04 DF6F 0120 SCAL SRLX ;Write to serial port.
2D06 18F8 0130 JR LOOP ;Do forever.
```

4.12 PROGRAM 12

Program 12 waits for a line of text to be entered from any of the devices in the input table, normally the keyboard. Following an Enter, the tape led is turned on and, after a delay, the content of the line on the screen is transmitted to the serial port. The tape led is then turned off and another input line awaited.

Listing of Program 12

```

          0010 ;PROGRAM 12
          0020 ;
2D00      0030      ORG 2D00H
          0040 ;
          0050 ;NAS-SYS routine numbers:
2D00 0063  0060 INLIN EQU 63H
2D00 005F  0070 MFLP  EQU 5FH
2D00 006D  0080 SOUT EQU 6DH
2D00 005D  0090 TDEL  EQU 5DH
          0100 ;
          0110 ;Get an input line:
2D00 DF63  0120 START SCAL INLIN
2D02 D5    0130      PUSH DE      ;Save address of line start.
          0140 ;Turn tape led on:
2D03 DF5F  0150      SCAL MFLP
          0160 ;Wait:
2D05 DF5D  0170      SCAL TDEL
          0180 ;Output the line to the serial port:
2D07 E1    0190      POP HL      ;Start of data.
2D08 0630  0200      LD B,30H    ;Line length is 48 chars.
2D0A DF6D  0210      SCAL SOUT
          0220 ;Turn tape led off.
2D0C DF5F  0230      SCAL MFLP
          0240 ;Do forever:
2D0E 18F0  0250      JR  START

```

4.13 PROGRAM 13

This program is a user-written output routine to drive a parallel printer. Such printers usually have seven or eight data lines which carry the character code to the printer, and two 'handshake' lines. One of these lines, STROBE, is pulsed low by the computer to transfer the character code into the printer register. While the printer is busy printing, it generates a level on its BUSY line to indicate to the computer that a further character code should not yet be transmitted.

The connections between the PIO and the printer are assumed to be:

PIO1D A0 -----> DATA 1
(Port 4)

A7 -----> DATA 8

PIO2D B0 -----> STROBE
(Port 6) B1 <----- BUSY

GND ----- GND

The program begins by initialising the two PIO ports to make PIO1 an 8 bit output port to carry the character code, and PIO2 a mode 3 port with bit 0 output and bits 1 to 7 input.

The STROBE signal is then set high, and the address of the PRINT subroutine loaded into workspace locations \$U0OUT+1/2. The initialisation section then returns to NAS-SYS.

The PRINT subroutine begins by waiting for the printer's BUSY signal to indicate that it is not busy. (It is assumed here that the printer indicates that it is busy by a logic 1 on its BUSY line; if the printer busy level is actually a logic 0, then the JR NZ,WAIT instruction must be replaced with a JR Z,WAIT instruction.)

If the output character is a carriage return it is replaced by the code for a linefeed. (If the printer automatically line feeds on receipt of a carriage return, lines 430 -450 inclusive must be removed.) The STROBE line is then brought low and then high again to strobe the character code into the printer.

To activate this parallel printer routine, execute it from OC80, ie command EC80; then, a U command will bring the printer routine into the table of output device. An N command removes the printer routine from the table. See U command, Chapter 2.

A Guide to NAS-SYS

Listins of Program 13

```

0010 ;PROGRAM 13
0020 ;
0030 ;User routine to drive a parallel printer.
0040 ;
OC80      0050      ORG 0C80H
0060 ;
0070 ;NAS-SYS routine numbers:
OC80 005B  0080 MRET EQU 5BH
0090 ;
0100 ;NAS-SYS workspace:
OC80 OC77  0110 $UOUT EQU 0C77H
0120 ;
0130 ;PIO definitions:
OC80 0004  0140 PAD EQU 4      ;Data, 8 bits.
OC80 0006  0150 PAC EQU PAD+2   ;Control register.
OC80 0005  0160 PBD EQU 5      ;Printer status.
OC80 0007  0170 PBC EQU PBD+2   ;Control register.
0180 ;Initialise PIOs:
OC80 3EOF  0190 LD A,0FH        ;PA,
OC82 D306  0200 OUT (PAC),A    ; output mode.
OC84 3ECF  0210 LD A,0CFH       ;PB,
OC86 D307  0220 OUT (PBC),A    ; mode 3...
OC88 3EFE  0230 LD A,0FEH       ;Bit 0 out,
OC8A D307  0240 OUT (PBC),A    ; others in.
0250 ;
0260 ;Set STROBE high:
OC8C 3E01  0270 LD A,1
OC8E D305  0280 OUT (PBD),A
0290 ;Initialise $UOUT:
OC90 21980C 0300 LD HL,PRINT
OC93 22780C 0310 LD ($UOUT+1),HL
0320 ;
OC96 DF5B  0330 SCAL MRET
0340 ;
0350 ;
OC98 F5   0360 PRINT PUSH AF    ;Save A.
OC99 DB05  0370 WAIT IN A,(PBD)  ;Wait for printer
OC9B CB4F  0380 BIT 1,A        ; not
OC9D 20FA  0390 JR NZ,WAIT     ; busy.
OC9F F1   0400 POP AF          ;Restore A.
OCA0 F5   0410 PUSH AF
OCA1 FE0D  0420 CP ODH         ;Is char a CR?
OCA3 2002  0430 JR NZ,CHAR     ;If not, then jump,
OCA5 3EOA  0440 LD A,0AH        ; else make it LF.
OCA7 D304  0450 CHAR OUT (PAD),A ;Output data.
OCA9 AF   0460 XOR A          ;Pull STROBE
OCAA D305  0470 OUT (PBD),A    ; low.
OCAC 00   0480 NOP
OCAD 3C   0490 INC A          ;STROBE high.
OCAE D305  0500 OUT (PBD),A
OCB0 F1   0510 POP AF
OCB1 C9   0520 RET

```

4.14 PROGRAM 14

This Program illustrates the use of the Relative CALL routine, RCAL. It simply prompts the user to input a single decimal digit and then displays that digit multiplied by three.

Observe that line 240, RCAL X3, produces machine code D7 04 which is the code for RST 10H : DEFB 04 where 04 is the displacement from the RCAL instruction to the required subroutine, ie 4025 - 401F - 2 = 04.

The assembled code occupies locations 4000H to 402CH and is executed from 4000H using the NAS-SYS command E4000. However, the point of using RCAL is that it allows position independent code to be written when used in conjunction with relative jumps, JR, as at line 270. Thus if the machine code is copied to another area of ram it will still operate. Try copying the code to location D00 onwards using NAS-SYS commands I4000 D00 2D followed by ED00.

Listing of Program 14

```

        0010 ;PROGRAM 14
        0020 ;
4000      0030          ORG 4000H
        0040 ;
        0050 ;NAS-SYS routine numbers
4000 0068  0060 B2HEX EQU 68H
4000 007B  0070 BLINK EQU 7BH
4000 006A  0080 CRLF EQU 6AH
4000 0028  0090 PRS EQU 28H
4000 0030  0100 ROUT EQU 30H
4000 0069  0110 SPACE EQU 69H
        0120 ;
        0130 ;Display prompt
4000 DF6A  0140 BEGIN SCAL CRLF
4002 EF    0150     RST PRS
4003 456E7465 0160     DEFM /Enter digit 0....9 /
72206469
67697420
302E2E2E
2E392020
4017 00    0170     DEFB 0
4018 DF7B  0180     SCAL BLINK      ;GET DIGIT
401A 47    0190     LD B,A          ;SAVE IT
401B F7    0200     RST ROUT        ;DISPLAY IT
401C DF69  0210     SCAL SPACE
401E 78    0220     LD A,B          ;RECOVER DIGIT
        0230 ;Now call routine to multiply
401F D704  0240     RCAL X3
        0250 ;Display result
4021 DF68  0260     SCAL B2HEX
4023 18DB  0270     JR BEGIN
        0280 ;

```

A Guide to NAS-SYS

Listing of Program 14, continued.

	0290	;	SUBROUTINE TO X 3.
	0300	;	ON ENTRY CHAR CODE IN A,
	0310	;	ON RETURN NUMBER IN A.
4025 D630	0320	X3	SUB 30H ;CONVERT TO NUMBER
4027 47	0330		LD B,A ;X1
4028 87	0340		ADD A,A ;X2
4029 27	0350		DAA ;CHANGE TO DECIMAL
402A 80	0360		ADD A,B ;X3
402B 27	0370		DAA ;CHANGE TO DECIMAL
402C C9	0380		RET